


The Little MLer



Matthias Felleisen and Daniel P. Friedman

Foreword by Robin Milner

© 1998 Massachusetts Institute of Technology

All rights reserved. No part of this book may be reproduced in any form by any electronic or mechanical means (including photocopying, recording, or information storage and retrieval) without permission in writing from the publisher.

This book was set by the authors and was printed and bound in the United States of America.

Library of Congress Cataloging-in-Publication Data

Felleisen, Matthias

The little MLer / Matthias Felleisen and Daniel P. Friedman; drawings by Duane Bibby; foreword by Robin Milner

p. cm.

Includes index and bibliographical references.

ISBN 0-262-56114-X (pbk : alk. paper)

1. ML (Computer program language) I. Friedman, Daniel P. II. Title.

QA76.73.M6F45 1998

005.13'3—dc21

97-40550

CIP

CONTENTS

Foreword ix

Preface xi

Experimenting with SML xiii

Experimenting with Objective Caml xv

1. Building Blocks 3

2. Matchmaker, Matchmaker 11

3. Cons Is Still Magnificent 31

4. Look to the Stars 45

5. Couples Are Magnificent, Too 57

6. Oh My, It's Full of Stars! 73

7. Functions Are People, Too 91

8. Bows and Arrows 109

9. Oh No! 133

10. Building on Blocks 151

Commencement 179

Index 180

FOREWORD

This is a book about writing programs, and understanding them as you write them. Most large computer programs are never completely understood; if they were, they wouldn't go wrong so often and we would be able to describe what they do in a scientific way. A good language should help to improve this state of affairs.

There are many ways of trying to understand programs. People often rely too much on one way, which is called “debugging” and consists of running a partly-understood program to see if it does what you expected. Another way, which ML advocates, is to install some means of understanding in the very programs themselves.

Standard ML was designed with this in mind. There are two particular ways-of-understanding built in to Standard ML; one is *types* for understanding data, the other is the *module system* for understanding the structure of the large-scale programs. People who program in a language with a strong type system, like this one, often say that their programs have fewer mistakes, and they understand them better.

The authors focus upon these features of Standard ML. They are well equipped to help you to understand programming; they are experienced teachers as well as researchers of the elegant and simple ideas which inspire good programming languages and good programming style.

Above all they have written a book which is a pleasure to read; it is free of heavy detail, but doesn't avoid tricky points. I hope you will enjoy the book and be able to use the ideas, whatever programming language you use in the future.

Robin Milner
Cambridge University

PREFACE

Programs consume data and produce data; designing a program requires a thorough understanding of data. In ML, programmers can express their understanding of the data using the sublanguage of types. Once the types are formulated, the design of the program follows naturally. Its shape will reflect the shape of the types and type definitions. Most collections of data, and hence most type specifications, are inductive, that is, they are defined in terms of themselves. Hence, most programs are recursive; again, they are defined in terms of themselves.

The first and primary goal of this book is to teach you to think recursively about types and programs. Perhaps the best programming language for understanding types and recursive thinking is ML. It has a rich, practical type language, and recursion is its natural computational mechanism. Since our primary concern is the idea of recursion, our treatment of ML in the first eight chapters is limited to the whys and wherefores of just a few features: types, datatypes, and functions.

The second goal of this book is to expose you to two important topics concerning large programs: dealing with exceptional situations and composing program components. Managing exceptional situations is possible, but awkward, with recursive functions. Consequently, ML provides a small and pragmatic sublanguage, *i.e.*, **exception**, **raise**, and **handle**, for dealing with such situations. The exception mechanism can also be used as a control tool to simplify recursive definitions when appropriate.

Typically, programs consist of many collections of many types and functions. Each collection is a program component or module. Constructing large programs means combining modules but also requires understanding the dependencies among the components. ML supports a powerful sublanguage for that purpose. In the last chapter, we introduce you to this language and the art of combining program components. The module sublanguage is again a functional programming language, just like the one we present in the first eight chapters, but its basic values are modules (called structures) not integers or booleans.

While *The Little MLer* provides an introduction to the principles of types, computation, and program construction, you should also know that ML itself is more general and incorporates more than we could intelligibly cover in an introductory text. After you have mastered this book, you can read and understand more advanced and comprehensive books on ML.

ACKNOWLEDGMENTS

We are indebted to Benjamin Pierce for numerous readings and insightful suggestions on improving the presentation and to Robert Harper for criticisms of the book and guidance concerning the new module system of ML. Michael Ashley, Cynthia Brown, Robby Findler, Matthew Flatt, Jeremy Frens, Steve Ganz, Daniel Grossman, Erik Hilsdale, Julia Lawall, Shinn-Der Lee, Michael Levin, David MacQueen, Kevin Millikin, Jon Riecke, George Springer, and Mitchell Wand read the book at various stages of development and their comments helped produce the final result. We also wish to thank Robert Prior at MIT Press who loyally supported us for many years. The book greatly benefited from Dorai Sitaram's incredibly clever Scheme typesetting program \LaTeX . Finally, we would like to thank the National Science Foundation for its continued support and especially for the Educational Innovation Grant that provided us with the opportunity to collaborate for the past year.

WHAT YOU NEED TO KNOW TO READ THIS BOOK

You must be comfortable reading English and performing rudimentary arithmetic. A willingness to use paper and pencil to ensure understanding is absolutely necessary.

READING GUIDELINES

Do not rush through this book. Read carefully; valuable hints are scattered throughout the text. Do not read the first eight chapters in fewer than three sittings. Allow one sitting at least for each of the last two chapters. Read systematically. If you do not *fully* understand one chapter, you will understand the next one even less.

The book is a dialogue about interesting examples of ML programs. If you can, try the examples while you read. Since ML implementations are predominantly interactive, the programmer can immediately participate in and observe the behavior of expressions. We encourage you to use this interactive read-evaluate-and-print loop to experiment with our definitions and examples. Some hints concerning experimentation are provided below.

We do not give any formal definitions in this book. We believe that you can form your own definitions and thus remember and understand them better than if we had written them out for you. But be sure you know and understand the morals that appear at the end of each chapter.

We use a few notational conventions throughout the text, primarily changes in typeface for different classes of symbols. Variables are in *italic*. Basic data, including numbers, booleans, constructors introduced via datatypes, are set in **sans serif**. Keywords, e.g., **datatype**, **of**, **and**, **fun**, are in **boldface**. When you experiment with the programs, you may ignore the typefaces but not the related framelines. To highlight this role of typefaces, the ML fragments in framelines are set in a **typewriter** face.

Food appears in many of our examples for two reasons. First, food is easier to visualize than abstract ideas. (This is not a good book to read while dieting.) We hope the choice of food will help you understand the examples and concepts we use. Second, we want to provide you with a little distraction. We know how frustrating the subject matter can be, and a little distraction will help you keep your sanity.

You are now ready to start. Good luck! We hope you will enjoy the experiences waiting for you on the following pages.

Bon appétit!

Matthias Felleisen
Daniel P. Friedman

EXPERIMENTING WITH SML

The book's programming language is a small subset of SML. With minor modifications, the examples of the first nine chapters of the book will run on most implementations of SML. For the tenth chapter, an implementation based on the 1996/97 revision of SML must be used.

The best mode to conduct experiments is

1. to place `Compiler.Control.Print.printDepth := 20;` into a newly created file,
2. to append the desired definitions (boxes) to the file,
3. to add a semicolon after each box, and
4. to employ `use "<filename>;` to load the definitions into the read-eval-print loop.

SML is then ready to accept and evaluate expressions that refer to the new definitions.

EXPERIMENTING WITH OBJECTIVE CAML

Objective Caml is a major dialect of the family of ML languages. The best mode to conduct experiments with Objective Caml is

1. to place `#print_depth 20;;` into a newly created file,
2. to append the desired definitions (boxes) to the file,
3. to add *two* semicolons after each box, and
4. to employ `#use "<filename>;"` to load the definitions into the read-eval-print loop.

Objective Caml is then ready to accept and evaluate expressions that refer to the new definitions.

Objective Caml's syntax differs slightly from SML's. By using the following hints systematically, you can easily translate the boxes from the first nine chapters of the book into Objective Caml. Each hint is marked by a chapter number and a frame number. If you are using Objective Caml, annotate the corresponding frames before you start reading to remind you where the differences between SML's and Objective Caml's syntaxes first appear.

1:16 Replace `datatype` by `type`:

```
type seasoning =  
  Salt  
  | Pepper
```

2:15 Replace `fun` by `let rec` and use `function`. The patterns omit the function name:

```
let rec only_onions =  
  function  
    (Skewer)  
    -> true  
  | (Onion(x))  
    -> only_onions(x)  
  | (Lamb(x))  
    -> false  
  | (Tomato(x))  
    -> false
```

4:66 To specify the precise types that a function should consume and produce, wrap the function name with the type assertion:

```
let rec (has_steak : meza * main * dessert -> bool) =  
  function  
    (x,Steak,d)  
    -> true  
  | (x,ns,d)  
    -> false
```


7:11 Since constructors are not functions in Objective Caml, define `hot_maker` as follows:

```
let rec hot_maker(x) =  
  function  
    (x)  
    -> Hot(x)
```

8:93 Curried definitions with matching on the first consumed value need parentheses around the function being returned when the second consumed value is placed in parentheses as we do here:

```
let rec combine_c =  
  function  
    (Empty)  
    -> (function  
        (12)  
        -> 12)  
  | (Cons(a,l1))  
    -> (function  
        (12)  
        -> Cons(a,combine_c(l1)(12)))
```

9:14 Replace `No_bacon 0` by `(No_bacon 0)`.

9:84 Replace `(exp1 handle pattern => exp2)` by `(try exp1 with pattern -> exp2)`.

Also, replace `div` by `/`:

```
let rec find(n,boxes) =  
  (try check(n,boxes,list_item(n,boxes))  
   with  
     Out_of_range  
     -> find(n / 2,boxes))  
and check =  
  function  
    (n,boxes,Bacon)  
    -> n  
  | (n,boxes,Ix(i))  
    -> find(i,boxes)
```

10 The examples of this chapter can be expressed in Objective Caml, but see the manual for the syntax of modules and simple examples that use them before you do.


The Little MLer

1. Building Blocks



Изображение, защищенное авторским правом

Изображение, защищенное авторским правом



Is this a number: 5?

¹ Yes.

Is 5 also an integer?

² Yes, it is.

Is this a number: 17?

³ Yes, it is also an integer.

Is this a number: -23?

⁴ Yes, but we don't use negative integers.

Is this an integer: 5.32?

⁵ No, and we don't use reals.

What type of number is 5?

⁶ *int*.¹

¹ The symbol *int* stands for "integer."

Quick, think of another integer!

⁷ How about 13?

What type of value is true?

⁸ *bool*.¹

¹ The symbol *bool* stands for "boolean."

What type of value is false?

⁹ *bool*.

Can you think of another *bool*?

¹⁰ No, that's all there is to *bool*.

Are there more *ints* than *bools*?

¹¹ Lots.

What is *int*?

¹² A type.

What is *bool*?

¹³ Another type.

What is a type?

¹⁴ A type is a name for a collection of values.

What is a type?

¹⁵ Sometimes we use it as if it were the collection.

Does this define a new type?

¹⁶ Yes, it does.

```
datatype seasoning =  
  Salt  
  | Pepper
```

Is this a *seasoning*: Salt?

¹⁷ Yes, it is.

And Pepper?

¹⁸ It's also a *seasoning*.

Can you think of another *seasoning*?

¹⁹ No, there are two. And that's all.

Have we seen a type like *seasoning* before?

²⁰ Yes, *bool* also has just two values.

Does this define a new type, too?

²¹ Yes, it does.

```
datatype num =  
  Zero  
  | One_more_than of num
```

Is this a *num*: Zero?

²² Obviously, just like Salt is a *seasoning*.

Is One_more_than(Zero) a *num*?

²³ Yes, because One_more_than constructs a *num* from a *num*.

How does `One_more_than` do that?

²⁴ We gave it `Zero`, which is a *num*, and it constructs a new *num*.

What is the type of
`One_more_than(
 One_more_than(
 Zero))`?

²⁵ *num*, because `One_more_than` constructs a *num* from a *num* and we agreed that
`One_more_than(
 Zero)`
is a *num*.

What is
`One_more_than(
 0)`?

²⁶ This is *nonsense*,¹ because `0` is not a *num*.

¹ We use the word “nonsense” for an expression that has no type.

What is the type of
`One_more_than(
 One_more_than(
 One_more_than(
 One_more_than(
 Zero))))`?

²⁷ *num*.

What is the difference between `Zero` and `0`?

²⁸ The value `Zero` belongs to the type *num*, whereas `0` belongs to *int*.

Correct. In general, if two things belong to two different types, they cannot be the same.

²⁹ A type is a name for a collection of values, and there is no overlap for any two distinct types.

Are there more *nums* than *bools*?

³⁰ Lots.

Are there more *nums* than *ints*?

³¹ No.¹

¹ And we will see in a later chapter why there are as many *ints* as *nums*.

What does this define?

```
datatype  $\alpha^1$  open_faced_sandwich =  
  Bread of  $\alpha$   
  | Slice of  $\alpha$  open_faced_sandwich
```

¹ We use 'a for α , but it is pronounced alpha.

³² It looks like the definition of a new type, but it also contains this funny looking α .

What is Bread(0)?

³³ It looks like an element of α *open_faced_sandwich*.

And what is Bread(true)?

³⁴ It also looks like an element of α *open_faced_sandwich*.
But how can both Bread(0) and Bread(true) be elements of the same type?

They can't! They belong to two different types:

int open_faced_sandwich

and

bool open_faced_sandwich.

³⁵ What does that mean?

It means that

```
datatype  $\alpha$  open_faced_sandwich =  
  Bread of  $\alpha$   
  | Slice of  $\alpha$  open_faced_sandwich
```

³⁶ Okay, that makes sense.

is not a type definition but a shape that represents many different datatypes.

So, if we write *int open_faced_sandwich*, we mean a type like this.¹

```
datatype int open_faced_sandwich =  
  Bread of int  
  | Slice of int open_faced_sandwich
```



What does *bool open_faced_sandwich* mean?

¹ The marker \otimes indicates that this definition is ungrammatical. We use this ungrammatical definition to explain α *open_faced_sandwich*.

³⁷ Writing *bool open_faced_sandwich* is as if we had defined a new **datatype**.

```
datatype bool open_faced_sandwich =  
  Bread of bool  
  | Slice of bool open_faced_sandwich
```



So what is *int open_faced_sandwich*?

³⁸ The simplest way of saying “This is an instance of the definition of α *open_faced_sandwich* where α stands for *int*.”

And what is *bool open_faced_sandwich*?

³⁹ The simplest way of saying “This is an instance of the definition of α *open_faced_sandwich* where α stands for *bool*.”

What is *num open_faced_sandwich*?

⁴⁰ The simplest way of saying “This is an instance of the definition of α *open_faced_sandwich* where α stands for *num*.”

Does that also mean that we can derive as many types as we want from the shape α *open_faced_sandwich*?

⁴¹ Yes.

Is

Bread(0)

an

int open_faced_sandwich?

⁴² Yes.

Why does it belong to

int open_faced_sandwich

and not

bool open_faced_sandwich?

⁴³ Because *0* is an *int*, and **Bread** constructs elements of type *int open_faced_sandwich* when it is given an *int*.

And what is the type of **Bread(true)**?

⁴⁴ *bool open_faced_sandwich*.

To what type does

**Bread(
 One_more_than(
 Zero))**

belong?

⁴⁵ It belongs to *num open_faced_sandwich*.

Is

Bread(Bread(0))

an

*(int open_faced_sandwich)
 open_faced_sandwich?*

⁴⁶ Yes, because *int open_faced_sandwich* is a type, and we said that we can derive a new type from α *open_faced_sandwich* by replacing α with any type.

And finally, since *(num open_faced_sandwich)* is also a type, to what type does

**Bread(
 Bread(
 One_more_than(
 Zero)))**

belong?

⁴⁷ It belongs to
 *(num open_faced_sandwich)
 open_faced_sandwich*.
Wow, types are types.

The First Moral

Use datatype to describe types. When a type contains lots of values, the datatype definition refers to itself. Use α with datatype to define shapes.

2. Matchmaker, Matchmaker



Imagem com direitos autorais

Here is another type definition.

¹ It contains four alternatives, not just two.

```
datatype shish_kebab =  
  Skewer  
| Onion of shish_kebab  
| Lamb of shish_kebab  
| Tomato of shish_kebab
```

What is different about it?

What is an element of this new type?

² How about
Skewer?

And another one?

³ Here's one:
Onion(
Skewer).

And a third?

⁴ Here's one more:
Onion(
Lamb(
Onion(
Skewer))).

Are there only Onions on this *shish_kebab*:
Skewer?

⁵ true, because there is neither Lamb nor
Tomato on the Skewer.

Are there only Onions on this *shish_kebab*:
Onion(
Skewer)?

⁶ true.

And how about:
Lamb(
Skewer)?

⁷ false, it contains Lamb.

Is it true that

```
Onion(  
  Onion(  
    Onion(  
      Skewer)))
```

contains only Onions?

⁸ true.

And finally:

```
Onion(  
  Lamb(  
    Onion(  
      Skewer)))?
```

⁹ false.

Is it true that

5

contains only Onions?

¹⁰ What kind of question is that? That looks like nonsense, because 5 is an *int*, not a *shish_kebab*.

Write the function *only_onions* using **fun**, **=**, **|**, **(,)**, **true**, **false**, **Skewer**, **Onion**, **Lamb**, and **Tomato**.

¹¹ Of course, you can't write this function, yet. Okay, you deserve something sweet for enduring this last question.

What kind of things does *only_onions* consume?

¹² *shish_kebabs*.

And what does it produce?

¹³ *bools*.

Are you anxious to see the first function definition?

¹⁴ Yes, we can't wait for the next page.

Here it is.

```
fun only_onions(Skewer)
  = true
| only_onions(Onion(x))
  = only_onions(x)
| only_onions(Lamb(x))
  = false
| only_onions(Tomato(x))
  = false
```

```
only_onions1 :
  shish_kebab → bool
```

Did you notice the second box?

¹ This box (type assertion) is a part of the program. It is transcribed as

```
(only_onions : shish_kebab → bool)
```

so that implementations can verify your thoughts about the type of a function. The transcription must always follow the function definition, never precede it. In general, if a box contains a bullet ●, then you must transcribe it by putting a left parenthesis in front of the contents and a right parenthesis behind it. The arrow is transcribed with two characters: - followed by >.

¹⁵ Yes, the second box is not a function definition. Why is the second box there?

The second box states what *only_onions* consumes and produces.

¹⁶ What is in front of (i.e., to the left of) the symbol → is the type of things that the function consumes, and what is behind → is the type of things it produces.

Is
 shish_kebab → *bool*
the type of *only_onions*?

¹⁷ Yes, *shish_kebab* → *bool* is the type of *only_onions* just as *int* is the type of 5.

Which item is mentioned first in the definition of *shish_kebab*?

¹⁸ Skewer.

Which item is mentioned first in the definition of *only_onions*?

¹⁹ Skewer.

Which item is mentioned second in the definition of *shish_kebab*?

²⁰ Onion.

Which item is mentioned second in the definition of *only_onions*?

²¹ Onion.

Does the sequence of items in the **datatype** definition correspond to the sequence in which they appear in the **function** definition?

²² Yes, it does. Is this always the case?

Almost always.

²³ Okay.

What is the value of
only_onions(
 Onion(
 Onion(
 Skewer))))?

²⁴ true.

And how do we determine the answer of
only_onions(
 Onion(
 Onion(
 Skewer))))?

²⁵ We will need to pay attention to the function definition.

```
fun only_onions(Skewer)
  = true
| only_onions(Onion(x))
  = only_onions(x)
| only_onions(Lamb(x))
  = false
| only_onions(Tomato(x))
  = false
```

Does
 only_onions(
 Onion(
 Onion(
 Skewer)))
match
 only_onions(Skewer)?

²⁶ No.

Why not?

²⁷ Because
Onion(
Onion(
Skewer))
does not match Skewer.

Does
only_onions(
Onion(
Onion(
Skewer)))
match
only_onions(Onion(*x*))?

²⁸ Yes, if *x* stands for
Onion(
Skewer).

Let *x* stand for
Onion(
Skewer).

²⁹ In that case, we have found a match.

Then what is
only_onions(
Onion(
Skewer))?

³⁰ It is
only_onions(*x*),
which is what follows the '=' below
only_onions(Onion(*x*)) in the definition of
only_onions, with *x* replaced by what it
stands for:
Onion(
Skewer).

Why do we need to know the meaning of
only_onions(
Onion(
Skewer))?

³¹ Because the answer for
only_onions(
Onion(
Skewer))
is also the answer for
only_onions(
Onion(
Onion(
Skewer))).

How do we determine the answer of

only_onions(
 Onion(
 Skewer))?

³² Let's see.

Does

only_onions(
 Onion(
 Skewer))

match

only_onions(Skewer)?

³³ No.

Why not?

³⁴ Because
 Onion(
 Skewer)
 does not match Skewer.

Does

only_onions(
 Onion(
 Skewer))

match

only_onions(Onion(*x*))?

³⁵ Yes, if *x* stands for Skewer, now.

So let *x* stand for Skewer, now.

³⁶ In that case, we have found our match again.

Then what is *only_onions*(Skewer)?

³⁷ It is
 only_onions(*x*),
 which is what follows the '=' below
 only_onions(Onion(*x*)) in the definition of
 only_onions, with *x* replaced by what it
 stands for:
 Skewer.

Why do we need to know what the meaning
of
 only_onions(Skewer)
is?

³⁸ Because the answer for
 only_onions(Skewer)
is the answer for
 only_onions(
 Onion(
 Skewer)),
which is the answer for
 only_onions(
 Onion(
 Onion(
 Skewer)))).

How do we determine the answer of
 only_onions(Skewer)?

³⁹ We need to match one more time.

Does
 only_onions(Skewer)
match
 only_onions(Skewer)?

⁴⁰ Completely.

Then what is the answer?

⁴¹ true.

Are we done?

⁴² Yes! The answer for
 only_onions(
 Onion(
 Onion(
 Skewer)))
is the same as the answer for
 only_onions(
 Onion(
 Skewer)),
which is the same as the answer for
 only_onions(Skewer),
which is
 true.

What is the answer of

```
only_onions(  
  Onion(  
    Lamb(  
      Skewer)))?
```

⁴³ false, isn't it?

Does

```
only_onions(  
  Onion(  
    Lamb(  
      Skewer)))
```

match

```
only_onions(Skewer)?
```

⁴⁴ No, it does not match.

Why not?

⁴⁵ Because
 Onion(
 Lamb(
 Skewer))
does not match Skewer.

Does

```
only_onions(  
  Onion(  
    Lamb(  
      Skewer)))
```

match

```
only_onions(Onion( $x$ ))?
```

⁴⁶ Yes, if x now stands for
 Lamb(
 Skewer).

Next let x stand for

```
Lamb(  
  Skewer).
```

⁴⁷ In that case, they match.

Then what is
 only_onions(
 Lamb(
 Skewer))?

⁴⁸ It is
 only_onions(*x*),
which is what follows the '=' below
 only_onions(Onion(*x*)), with *x* replaced by
what it stands for:
 Lamb(
 Skewer).

Why do we need to know what
 only_onions(
 Lamb(
 Skewer))
is?

⁴⁹ Because the answer for
 only_onions(
 Lamb(
 Skewer))
is the answer for
 only_onions(
 Onion(
 Lamb(
 Skewer)))).

Does
 only_onions(
 Lamb(
 Skewer))
match
 only_onions(Skewer)?

⁵⁰ No.

Does
 only_onions(
 Lamb(
 Skewer))
match
 only_onions(Onion(*x*))?

⁵¹ No.

Does
 only_onions(
 Lamb(
 Skewer))
match
 only_onions(Lamb(*x*))?

⁵² Yes, if *x* stands for Skewer, now

And now what is the answer?

⁵³ false, because false follows the '=' below *only_onions*(Lamb(*x*)) in the definition of *only_onions*.

Are we done?

⁵⁴ Yes! The answer for
only_onions(
 Onion(
 Lamb(
 Skewer)))
is the same as the answer for
only_onions(
 Lamb(
 Skewer)),
which is
false.

Describe the function *only_onions* in your own words.

⁵⁵ Here are our words:
“*only_onions* consumes a *shish_kebab* and checks to see whether it is only edible by an onion lover.”

Describe how the function *only_onions* accomplishes this.

⁵⁶ Here are our words again:
“*only_onions* looks at each piece of the *shish_kebab* and, if it doesn't encounter Lamb or Tomato, it produces true.”

So what is the value of
only_onions(5)?

⁵⁷ Nonsense. We already said that 5 is an *int*, not a *shish_kebab*.

Is
 Tomato(
 Skewer)?
an element of *shish_kebab*?

⁵⁸ Yes.

Is
 Onion(
 Tomato(
 Skewer))
an element of *shish_kebab*?

⁵⁹ Since
 Tomato(
 Skewer)
is an element of *shish_kebab*, we can also
wrap an Onion around it.

And how about another Tomato?

⁶⁰ Sure.

Is
 Tomato(
 Onion(
 Tomato(
 Skewer)))
a vegetarian shish kebab?

⁶¹ Of course, there is no Lamb in it.

Is
 Onion(
 Onion(
 Onion(
 Skewer)))
a vegetarian shish kebab?

⁶² Yes, it only contains Onions.

Define the function
 is_vegetarian :
 shish_kebab → *bool*,
which returns true if what it consumes does
not contain Lamb.

⁶³ Shouldn't the line for Tomatoes in this
function be the same as the line for Onions?

```
fun is_vegetarian(Skewer)
  = true
| is_vegetarian(Onion(x))
  = is_vegetarian(x)
| is_vegetarian(Lamb(x))
  = false
| is_vegetarian(Tomato(x))
  = is_vegetarian(x)
```

```
is_vegetarian :
  shish_kebab → bool
```

•

Yes, that's right. Let's move on. What does

⁶⁴ It defines a datatype that is similar in shape to *shish_kebab*.

```
datatype  $\alpha$  shish =  
  Bottom of  $\alpha$   
| Onion of  $\alpha$  shish  
| Lamb of  $\alpha$  shish  
| Tomato of  $\alpha$  shish
```

define?

Do the definitions of α *shish* and *shish_kebab* use the same names?

⁶⁵ Yes, the names of the constructors are the same, but clearly from now on Onion constructs an α *shish* and no longer a *shish_kebab*.

What is different about the new datatype?

⁶⁶ A *shish_kebab* is always on a Skewer, an α *shish* is placed on different kinds of Bottoms.

Here are some bottom objects.

⁶⁷ Sure, *rod shish* makes some form of shish kebab.

```
datatype rod =  
  Dagger  
| Fork  
| Sword
```

Are they good ones?

Think of another class of bottom objects.

⁶⁸ We could move all of the food to various forms of plates.

```
datatype plate =  
  Gold_plate  
| Silver_plate  
| Brass_plate
```

What is the type of

```
Onion(  
  Tomato(  
    Bottom(Dagger)))?
```

⁶⁹ It belongs to *rod shish*.

Is
 Onion(
 Tomato(
 Bottom(Dagger)))
a vegetarian *rod shish*?

⁷⁰ Sure it is. It only contains Tomatoes and Onions.

Does
 Onion(
 Tomato(
 Bottom(Gold_plate)))
belong to *plate shish*?

⁷¹ Sure, because Gold_plate is a *plate* and *plate* is used as a Bottom, and Tomatoes and Onions can be wrapped around Bottoms.

Is
 Onion(
 Tomato(
 Bottom(Gold_plate)))
a vegetarian shish kebab?

⁷² Sure it is. It is basically like
 Onion(
 Tomato(
 Bottom(Dagger)))
except that we have moved all the food from a Dagger to a Gold_plate.

Let's define the function
 $is_veggie : \alpha \text{ shish} \rightarrow bool$,
which checks whether a shish kebab contains only vegetarian foods, regardless of what Bottom it is in.

⁷³ It only differs from *is_vegetarian* in one part.

```
fun is_veggie(Bottom(x))  
  = true  
| is_veggie(Onion(x))  
  = is_veggie(x)  
| is_veggie(Lamb(x))  
  = false  
| is_veggie(Tomato(x))  
  = is_veggie(x)
```

```
is_veggie :  
   $\alpha \text{ shish} \rightarrow bool$ 
```

•

This new function matches against arbitrary Bottoms, whereas *is_vegetarian* only matches against Skewers.

Let's determine the value of

```
is_veggie(  
  Onion(  
    Fork)).
```

⁷⁴ This is nonsense.

Why?

⁷⁵ Because `Onion` constructs α *shish* from α *shish*, which does not include `Fork`.

What is the value of

```
is_veggie(  
  Onion(  
    Tomato(  
      Bottom(Dagger))))?
```

⁷⁶ `true`.

What type of thing is

```
Onion(  
  Tomato(  
    Bottom(Dagger)))?
```

⁷⁷ We said it belonged to the type *rod shish*.

What is the value of

```
is_veggie(  
  Onion(  
    Tomato(  
      Bottom(Gold_plate))))?
```

⁷⁸ It is `true`, too.

And what type of thing is

```
Onion(  
  Tomato(  
    Bottom(Gold_plate)))?
```

⁷⁹ It belongs to the type *plate shish*, which has the same shape as *rod shish*, but is a distinct type.

But aren't both examples of α *shish*?

⁸⁰ Yes, they are. The two types only differ in how α is replaced by a type.

How can `is_veggie` consume things that belong to different types?

⁸¹ Perhaps we should think of `is_veggie` as two functions.

What functions should we think about?

⁸² One function has the type
 $rod\ shish \rightarrow bool$
and the other one has the type
 $plate\ shish \rightarrow bool$.

Where else do the functions differ?

⁸³ Nowhere—they are identical otherwise.

So this is how we could have written the function *is_veggie* for *shishes* on rods.

```
datatype rod =  
  Dagger  
| Fork  
| Sword
```

```
fun is_veggie(Bottom(x))  
  = true  
| is_veggie(Onion(x))  
  = is_veggie(x)  
| is_veggie(Lamb(x))  
  = false  
| is_veggie(Tomato(x))  
  = is_veggie(x)
```

```
is_veggie :  
  rod shish  $\rightarrow$  bool
```

```
datatype plate =  
  Gold_plate  
| Silver_plate  
| Brass_plate
```

```
fun is_veggie(Bottom(x))  
  = true  
| is_veggie(Onion(x))  
  = is_veggie(x)  
| is_veggie(Lamb(x))  
  = false  
| is_veggie(Tomato(x))  
  = is_veggie(x)
```

```
is_veggie :  
  plate shish  $\rightarrow$  bool
```

And how would we write the function *is_veggie* for *shishes* on plates?

Whew, that's a lot of writing!

What type of value is

```
is_veggie(  
  Onion(  
    Tomato(  
      Bottom(52))))?)
```

⁸⁵ *bool*.

What type of value is ⁸⁶ *bool*.

```
is_veggie(  
  Onion(  
    Tomato(  
      Bottom(  
        One_more_than(Zero))))))?
```

What type of value is ⁸⁷ *bool*.

```
is_veggie(  
  Onion(  
    Tomato(  
      Bottom(false))))?
```

Does that mean *is_veggie* works for all five types: *rod shish*, *plate shish*, *int shish*, *num shish*, and *bool shish*? ⁸⁸ Yes, and all other *shish* types that we could possibly think of.

What is the bottom object of ⁸⁹ All the food is on a dagger.

```
Onion(  
  Tomato(  
    Bottom(Dagger)))?
```

What is the bottom object of ⁹⁰ All the food is now on a gold plate.

```
Onion(  
  Tomato(  
    Bottom(Gold_plate)))?
```

What is the bottom object of ⁹¹ All the food is on a 52.

```
Onion(  
  Tomato(  
    Bottom(52)))?
```

What is the value of ⁹² Dagger.

```
what_bottom(  
  Onion(  
    Tomato(  
      Bottom(Dagger))))?
```

What is the value of
 `what_bottom(
 Onion(
 Tomato(
 Bottom(Gold_plate))))`?

⁹³ Gold_plate.

What is the value of
 `what_bottom(
 Onion(
 Tomato(
 Bottom(52))))`?

⁹⁴ 52.

So what type of value does *what_bottom* consume?

⁹⁵ α *shish*, which means all types of *shishes*.

And what type of value does *what_bottom* produce?

⁹⁶ It produces *rods*, *plates*, and *ints*. And it looks like it can produce a whole lot more.

Is there a simple way of saying what type of value it produces?

⁹⁷ Here is our way:
 “If α is a type and we use *what_bottom* on a value of type α *shish*, then the result is of type α .”

How many variants of *shishes* must *what_bottom* match?

⁹⁸ There are four.

```
fun what_bottom(Bottom(x))
  = _____
  | what_bottom(Onion(x))
  = _____
  | what_bottom(Lamb(x))
  = _____
  | what_bottom(Tomato(x))
  = _____
```

What is the value of
 `what_bottom(
 Bottom(52))`?

⁹⁹ 52.

What is the value of ¹⁰⁰ `Sword`.
`what_bottom(
 Bottom(Sword))`?

What is the value of ¹⁰¹ `x`.
`what_bottom(
 Bottom(x))`,
no matter what `x` is?

So what goes into the first blank line of ¹⁰² `x`.
`what_bottom`?

What is the value of ¹⁰³ `52`.
`what_bottom(
 Tomato(
 Onion(
 Lamb(
 Bottom(52))))))`?

What is the value of ¹⁰⁴ `52`.
`what_bottom(
 Onion(
 Lamb(
 Bottom(52))))`?

What is the value of ¹⁰⁵ `52`.
`what_bottom(
 Lamb(
 Bottom(52)))`?

What is the value of ¹⁰⁶ `52`.
`what_bottom(
 Bottom(52))`?

Does that mean that the value of

```
what_bottom(  
  Tomato(  
    Onion(  
      Lamb(  
        Bottom(52))))))
```

is the same as

```
what_bottom(  
  Onion(  
    Lamb(  
      Bottom(52))))),
```

which is the same as

```
what_bottom(  
  Lamb(  
    Bottom(52))),
```

which is the same as

```
what_bottom(  
  Bottom(52))?
```

¹⁰⁷ Yes, all four have the same answer: 52.

Fill in the blanks in this skeleton.

```
fun what_bottom(Bottom(x))  
  = x  
| what_bottom(Onion(x))  
  = what_bottom(x)  
| what_bottom(Lamb(x))  
  = _____  
| what_bottom(Tomato(x))  
  = _____
```

¹⁰⁸ Now this is easy.

```
fun what_bottom(Bottom(x))  
  = x  
| what_bottom(Onion(x))  
  = what_bottom(x)  
| what_bottom(Lamb(x))  
  = what_bottom(x)  
| what_bottom(Tomato(x))  
  = what_bottom(x)
```


```
what_bottom :  
   $\alpha$  shish  $\rightarrow \alpha$ 
```

•

The Second Moral

The number and order of the patterns in the definition of a function should match that of the definition of the consumed datatype.

3. Cons Is Still Magnificent



Do you like to eat pizza?

¹ Looks like good toppings.

```
datatype pizza =  
  Crust  
  | Cheese of pizza  
  | Onion of pizza  
  | Anchovy of pizza  
  | Sausage of pizza
```

Here is our favorite pizza:

² This looks too salty.

```
Anchovy(  
  Onion(  
    Anchovy(  
      Anchovy(  
        Cheese(  
          Crust)))))).
```

How about removing each Anchovy?

³ That would make it less salty.

Let's remove them. What is the value of

```
remove_anchovy(  
  Anchovy(  
    Onion(  
      Anchovy(  
        Anchovy(  
          Cheese(  
            Crust))))))?)
```

⁴ It should be a Cheese and Onion *pizza*, like this:

```
Onion(  
  Cheese(  
    Crust)).
```

What is the value of

```
remove_anchovy(  
  Sausage(  
    Onion(  
      Anchovy(  
        Sausage(  
          Cheese(  
            Crust))))))?)
```

⁵ It should be a Cheese, Sausage, and Onion *pizza*, like this:

```
Sausage(  
  Onion(  
    Sausage(  
      Cheese(  
        Crust)))).
```

Does *remove_anchovy* consume pizzas?

⁶ Yes, and it produces them, too.

Fill in the blanks in the skeleton.

⁷ We didn't expect you to know this one.

```
fun remove_anchovy(Crust)
  = Crust
  | remove_anchovy(Cheese(x))
  = _____
  | remove_anchovy(Onion(x))
  = _____
  | remove_anchovy(Anchovy(x))
  = _____
  | remove_anchovy(Sausage(x))
  = _____
```

```
remove_anchovy :
  pizza → pizza
```

Fill in all the blanks except for the **Anchovy** line.

⁸ The **Onion** and **Sausage** lines are similar to the **Cheese** line.

```
fun remove_anchovy(Crust)
  = Crust
  | remove_anchovy(Cheese(x))
  = Cheese(remove_anchovy(x))
  | remove_anchovy(Onion(x))
  = _____
  | remove_anchovy(Anchovy(x))
  = _____
  | remove_anchovy(Sausage(x))
  = _____
```

```
fun remove_anchovy(Crust)
  = Crust
  | remove_anchovy(Cheese(x))
  = Cheese(remove_anchovy(x))
  | remove_anchovy(Onion(x))
  = Onion(remove_anchovy(x))
  | remove_anchovy(Anchovy(x))
  = _____
  | remove_anchovy(Sausage(x))
  = Sausage(remove_anchovy(x))
```

We've eaten the cheese already.

Explain why we use **Cheese**, **Onion**, and **Sausage** when we fill in the blanks.

⁹ For every **Cheese**, **Onion**, or **Sausage** that we see, we must put one back.

Since *remove_anchovy* must produce a *pizza*, let us use *Crust*, the simplest *pizza*, for the line that contains *Anchovy*(*x*).

```
fun remove_anchovy(Crust)
  = Crust
  | remove_anchovy(Cheese(x))
    = Cheese(remove_anchovy(x))
  | remove_anchovy(Onion(x))
    = Onion(remove_anchovy(x))
  | remove_anchovy(Anchovy(x))
    = Crust
  | remove_anchovy(Sausage(x))
    = Sausage(remove_anchovy(x))
```

¹⁰ Yes, *remove_anchovy* consumes *pizza* and produces *pizza* without *Anchovy* on it.

Let's try it out on a small pizza:

```
remove_anchovy(
  Anchovy(
    Crust)).
```

¹¹ That's easy. It matches the *Anchovy* line, if *x* stands for *Crust*. And the answer is *Crust*.

Is

Crust

like

```
remove_anchovy(
  Anchovy(
    Crust))
```

without *Anchovy*?

¹² Absolutely, but what if we had more anchovies?

No problem. Here is an example:

```
remove_anchovy(
  Anchovy(
    Anchovy(
      Crust))).
```

¹³ That's easy again. It also matches the *Anchovy* line and the answer is still *Crust*.

Okay, so what if we had onions on top:

```
remove_anchovy(  
  Onion(  
    Cheese(  
      Anchovy(  
        Anchovy(  
          Crust))))))?
```

¹⁴ This matches
remove_anchovy(Onion(*x*))
if *x* stands for
Cheese(
 Anchovy(
 Anchovy(
 Crust))).

What is the value of
Onion(remove_anchovy(x))

if *x* stands for
Cheese(
 Anchovy(
 Anchovy(
 Crust))))?

¹⁵ It is the pizza that
remove_anchovy(
 Cheese(
 Anchovy(
 Anchovy(
 Crust))))
produces, with Onion added on top.

What is the value of

```
remove_anchovy(  
  Cheese(  
    Anchovy(  
      Anchovy(  
        Crust)))))?
```

¹⁶ This matches
remove_anchovy(Cheese(*x*))
if *x* stands for
Anchovy(
 Anchovy(
 Crust)).

And what is the value of

```
Cheese(remove_anchovy(x))
```

if *x* stands for
Anchovy(
 Anchovy(
 Crust))))?

¹⁷ It is the pizza that
remove_anchovy(
 Anchovy(
 Anchovy(
 Crust))))
produces, with Cheese added on top.

Do we know the value of

```
remove_anchovy(  
  Anchovy(  
    Anchovy(  
      Crust))))?
```

¹⁸ Yes, we know that this produces Crust.

Does that mean that `Crust` is the answer?

¹⁹ No, we still have to add `Cheese` and `Onion`.

Does it matter in which order we add those two ingredients?

²⁰ Yes, we must first add `Cheese`, producing
 `Cheese(
 Crust)`
and then we add `Onion`.

So what is the final answer?

²¹ It is
 `Onion(
 Cheese(
 Crust))`.

Can you describe in your own words what `remove_anchovy` does?

²² Here are our words:
 “`remove_anchovy` looks at each topping of a *pizza* and makes a pizza with all the toppings that are above the first anchovy.”

Is that what we wanted?

²³ No. We wanted to keep all toppings except for anchovies.

Let's try one more example:

```
remove_anchovy(  
  Cheese(  
    Anchovy(  
      Cheese(  
        Crust))))).
```

²⁴ It should be a double-cheese pizza.

What kind of pizza should this make?

Check it out!

²⁵ It matches
 `remove_anchovy(Cheese(x))`
if *x* stands for
 `Anchovy(
 Cheese(
 Crust))`.

Doesn't that mean that the result is

```
Cheese(  
  remove_anchovy(  
    Anchovy(  
      Cheese(  
        Crust)))))?
```

²⁶ Yes, we have at least one Cheese topping.

What does

```
remove_anchovy(  
  Anchovy(  
    Cheese(  
      Crust)))
```

match next?

²⁷ This matches
`remove_anchovy(Anchovy(x)).`

And the answer is
Crust?

²⁸ Yes, and just like before we need to add
Cheese on top.

Does that mean the final answer is
Cheese(
 Crust)?

²⁹ Yes, but that's not the answer we wanted.

What did we want?

³⁰ A double-cheese pizza like
Cheese(
 Cheese(
 Crust)),
because that's what it means to remove
anchovies and nothing else.

How do we have to change *remove_anchovy* to get the Cheese back?

³¹ The Anchovy line must produce *remove_anchovy(x)*.

```
fun remove_anchovy(Crust)
  = Crust
  | remove_anchovy(Cheese(x))
  = Cheese(remove_anchovy(x))
  | remove_anchovy(Onion(x))
  = Onion(remove_anchovy(x))
  | remove_anchovy(Anchovy(x))
  = remove_anchovy(x)
  | remove_anchovy(Sausage(x))
  = Sausage(remove_anchovy(x))
```

Does this new version of *remove_anchovy* still consume pizzas?

³² Yes, and it still produces them.

You have earned yourself a double-cheese pizza.

³³ And don't forget the anchovies.

Would you like even more cheese than that?

³⁴ Some people like lots of cheese.

We could add cheese on top of the anchovies.

³⁵ Yes, that would hide their taste a bit.

What kind of pizza is
top_anchovy_with_cheese(
 Onion(
 Anchovy(
 Cheese(
 Anchovy(
 Crust))))))?)

³⁶ Easy, there is a layer of Cheese on top of each Anchovy:
 Onion(
 Cheese(
 Anchovy(
 Cheese(
 Cheese(
 Anchovy(
 Crust)))))))).

And what is

```
top_anchovy_with_cheese(  
  Onion(  
    Cheese(  
      Sausage(  
        Crust)))))?
```

³⁷ Here we don't add any Cheese, because the pizza does not contain any Anchovy:

```
Onion(  
  Cheese(  
    Sausage(  
      Crust))).
```

Fill in the blanks in the skeleton.

```
fun top_anchovy_with_cheese(Crust)  
  = Crust  
  | top_anchovy_with_cheese(Cheese(x))  
  = _____  
  | top_anchovy_with_cheese(Onion(x))  
  = _____  
  | top_anchovy_with_cheese(Anchovy(x))  
  = _____  
  | top_anchovy_with_cheese(Sausage(x))  
  = _____
```

```
top_anchovy_with_cheese :  
  pizza → pizza
```

³⁸ We expect you to know some of the answers.

```
fun top_anchovy_with_cheese(Crust)  
  = Crust  
  | top_anchovy_with_cheese(Cheese(x))  
  = Cheese(top_anchovy_with_cheese(x))  
  | top_anchovy_with_cheese(Onion(x))  
  = Onion(top_anchovy_with_cheese(x))  
  | top_anchovy_with_cheese(Anchovy(x))  
  = _____  
  | top_anchovy_with_cheese(Sausage(x))  
  = Sausage(top_anchovy_with_cheese(x))
```

How does that skeleton compare with this one?

```
fun remove_anchovy(Crust)  
  = Crust  
  | remove_anchovy(Cheese(x))  
  = Cheese(remove_anchovy(x))  
  | remove_anchovy(Onion(x))  
  = Onion(remove_anchovy(x))  
  | remove_anchovy(Anchovy(x))  
  = _____  
  | remove_anchovy(Sausage(x))  
  = Sausage(remove_anchovy(x))
```

³⁹ The two skeletons are the same except for the names of the functions.

What function would we get if we filled the blank in the last skeleton for

top_anchovy_with_cheese

with

top_anchovy_with_cheese(x)?

⁴⁰ We would get *remove_anchovy* but with a different name.

Then what do we have to put into the blank? ⁴¹

We must at least put the Anchovy back on the pizza.

And then?

⁴² We must top it with Cheese.

Let's do it!

⁴³ Here it is.

```
fun top_anchovy_with_cheese(Crust)
  = Crust
  | top_anchovy_with_cheese(Cheese(x))
    = Cheese(top_anchovy_with_cheese(x))
  | top_anchovy_with_cheese(Onion(x))
    = Onion(top_anchovy_with_cheese(x))
  | top_anchovy_with_cheese(Anchovy(x))
    = Cheese(
      Anchovy(
        top_anchovy_with_cheese(x)))
  | top_anchovy_with_cheese(Sausage(x))
    = Sausage(top_anchovy_with_cheese(x))
```

What type of value does

top_anchovy_with_cheese

produce?

⁴⁴ The difference between
top_anchovy_with_cheese
and

remove_anchovy

is one line. Cheese on top of Anchovy on a pizza still makes *pizza*, so the type of

top_anchovy_with_cheese

is

pizza \rightarrow *pizza*.

How many occurrences of **Cheese** are in the result of

```
top_anchovy_with_cheese(  
  remove_anchovy(  
    Onion(  
      Anchovy(  
        Cheese(  
          Anchovy(  
            Crust))))))?)
```

⁴⁵ One, because *remove_anchovy* removes all anchovies, so that *top_anchovy_with_cheese* doesn't add any cheese.

How many occurrences of **Cheese** are in the result of

```
remove_anchovy(  
  top_anchovy_with_cheese(  
    Onion(  
      Anchovy(  
        Cheese(  
          Anchovy(  
            Crust))))))?)
```

⁴⁶ Three, because *top_anchovy_with_cheese* first adds **Cheese** for each **Anchovy**. Then *remove_anchovy* removes all anchovies:

```
Onion(  
  Cheese(  
    Cheese(  
      Cheese(  
        Crust))))).
```

Perhaps we should replace every **Anchovy** with **Cheese**.

⁴⁷ We just did that for one pizza.

Is it true that for each **Anchovy** in x

```
remove_anchovy(  
  top_anchovy_with_cheese( $x$ ))
```

adds some **Cheese** as long as x is a pizza?

⁴⁸ Yes, and it does more. Once all the cheese is added, the anchovies are removed.

So is this the correct definition of *subst_anchovy_by_cheese*?

⁴⁹ Yes, it is. This function replaces each instance of **Anchovy** by **Cheese**.

```
fun subst_anchovy_by_cheese( $x$ )  
  = remove_anchovy(  
    top_anchovy_with_cheese( $x$ ))
```

```
subst_anchovy_by_cheese :  
  pizza  $\rightarrow$  pizza
```

•

Can you describe in your own words how
subst_anchovy_by_cheese
works?

⁵⁰ Here are our words:
“*subst_anchovy_by_cheese* looks at each
topping of a pizza and adds Cheese on top
of each Anchovy. Then, it looks at each
topping again, including all the new
cheese, and removes the anchovies.”

Here are some different words:
“*subst_anchovy_by_cheese* looks at each
topping of a pizza and replaces each
Anchovy by Cheese.”
Can you define a function that matches this
description and doesn’t use *remove_anchovy*
and *top_anchovy_with_cheese*?

⁵¹ Yes, here is a skeleton.

```
fun subst_anchovy_by_cheese(Crust)
  = Crust
  | subst_anchovy_by_cheese(Cheese(x))
  = Cheese(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Onion(x))
  = Onion(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Anchovy(x))
  = _____
  | subst_anchovy_by_cheese(Sausage(x))
  = Sausage(subst_anchovy_by_cheese(x))
```

Does this skeleton look familiar?

⁵² Yes, this skeleton looks just like those of
top_anchovy_with_cheese
and
remove_anchovy.

Fill in the blank.

⁵³ Here it is.

```
fun subst_anchovy_by_cheese(Crust)
  = Crust
  | subst_anchovy_by_cheese(Cheese(x))
  = Cheese(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Onion(x))
  = Onion(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Anchovy(x))
  = _____
  | subst_anchovy_by_cheese(Sausage(x))
  = Sausage(subst_anchovy_by_cheese(x))
```

```
fun subst_anchovy_by_cheese(Crust)
  = Crust
  | subst_anchovy_by_cheese(Cheese(x))
  = Cheese(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Onion(x))
  = Onion(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Anchovy(x))
  = Cheese(subst_anchovy_by_cheese(x))
  | subst_anchovy_by_cheese(Sausage(x))
  = Sausage(subst_anchovy_by_cheese(x))
```

Now you can replace Anchovy with whatever ⁵⁴ We will stick with anchovies.
pizza topping you want.

The Third Moral

Functions that produce values of a datatype must use the associated constructors to build data of that type.

4. Look to the Stars



Are you tired of making pizza?

¹ We are too. Let's make complete meals.

Do you like shrimp cocktail?

² We do, too.

We like Hummus for meza too.

³ And how about some Escargots?

Okay, let's sum them up.

⁴ There is a new one, too: Calamari.

```
datatype meza =  
  Shrimp  
  | Calamari  
  | Escargots  
  | Hummus
```

And here are some entrées.

⁵ We should also have some salads.

```
datatype main =  
  Steak  
  | Ravioli  
  | Chicken  
  | Eggplant
```

```
datatype salad =  
  Green  
  | Cucumber  
  | Greek
```

Let's not forget the fun part.

⁶ Yes, we need desserts.

```
datatype dessert =  
  Sundae  
  | Mousse  
  | Torte
```

Now let's make a meal.

⁷ Don't we have to put together different courses when we make full meals?

No, we can use stars!

⁸ What is a star?

Here is our first three star meal:
(Calamari,Ravioli,Greek,Sundae).

⁹ It looks like a meal.

How many items does this meal have?

¹⁰ Four, and they are separated by commas and enclosed in parentheses.

Is
(Hummus,Steak,Green,Torte)
a meal of the same type?

¹¹ Yes, it also consists of four items in the same order: *meza*, *main*, *salad*, and *dessert*.

Does
(Torte,Hummus,Steak,Sundae)
belong to the same type?

¹² We have seen meals like this before, but *dessert* should never be the first course.

The first kind of meal is of type
(*meza * main * salad * dessert*).

¹³ Does this mean that the type of the thing that is not a meal is
(*dessert * meza * main * dessert*)?

What's unusual about our meals?

¹⁴ People here eat the salads before the main course:
(*meza * salad * main * dessert*).

Is that a meal?

¹⁵ It is not the same kind of meal, is it?

No, it is not. Each star corresponds to a comma in the construction of a meal.

¹⁶ And the order matters, right?

Yes, the order matters, but do we have to have three stars in meals?

¹⁷ No, if we want small meals with three courses, we only need two stars. And if we want tiny meals with two courses, we need only one.

What is your favorite kind of meal with only two ingredients?

¹⁸ Ours is
(Shrimp,Sundae).

What is the type of that tiny meal?

¹⁹ (*meza* * *dessert*).

Have you tasted your sundae yet?

²⁰ We just ate ours.

What is
add_a_steak(Shrimp)?

²¹ It is a tiny meal:
(Shrimp,Steak).

What is
add_a_steak(Hummus)?

²² This meal needs something to sink our teeth into.
(Hummus,Steak).

Does *add_a_steak* consume *meza*?

²³ Yes, it does.

Does *add_a_steak* produce a tiny meal?

²⁴ Yes, this function always produces a tiny meal. Indeed, we even know that the second item is always **Steak**.

Is this a definition of *add_a_steak*?

²⁵ It is a function and we already discussed what it consumes and produces.

```
fun add_a_steak(Shrimp)
  = (Shrimp,Steak)
| add_a_steak(Calamari)
  = (Calamari,Steak)
| add_a_steak(Escargots)
  = (Escargots,Steak)
| add_a_steak(Hummus)
  = (Hummus,Steak)
```

```
add_a_steak :
  meza → (meza * main)
```

•

What is its type?

Isn't this long for something so simple?

²⁶ Yes, four lines is a lot. Can we shorten it?

It doesn't really matter what the *meza* is, so we can just give it a name in the pattern and use that name in the answer. Define the abridged version of *add_a_steak*.

²⁷ With this hint, it is a piece of cake (which, by the way, isn't a *dessert*).

```
fun add_a_steak(x)
  = (x,Steak)
```

What is the value of
add_a_steak(Escargots)?

²⁸ (Escargots,Steak).

And how about *add_a_steak*(5)?

²⁹ Isn't this nonsense?

It should be.

³⁰ But is it?

It would be nonsense had we only used the first version of *add_a_steak*.

³¹ Correct. It consumed only *meza*.

What does the abridged version of *add_a_steak* consume?

³² Anything.

So what is its type?

³³ We have always used α when a function could consume anything.

```
add_a_steak :
   $\alpha \rightarrow (\alpha * \text{main})$ 
```

•

Does that mean the second version of *add_a_steak* is more general than the first?

³⁴ Yes, the second version exists for many different types. Therefore it can consume *mezas*, or *desserts*, or *nums*, and even *mains*.

Are both definitions correct?

³⁵ Yes, they both add a **Steak**.

Why should we choose one over the other?

³⁶ We know that the second one is more general, but it is also always one line long. The first kind of definition always contains as many lines as there are alternatives in the **datatype** definition.

Is it always better to use the more general version?

³⁷ No, the more specific one is more accurate, so using it will reveal nonsense more often.

Could we have used this idea of shortening functions before?

³⁸ Yes, we should have known about this shorthand when we defined *remove_anchovy*. It could have been so much shorter.

```
fun remove_anchovy(Crust)
  = Crust
  | remove_anchovy(Anchovy(x))
  = remove_anchovy(x)
  | remove_anchovy(C(x))
  = C(remove_anchovy(x))
```



Nice dream, but it is impossible for a variable like *C* to stand in place of a constructor that consumes values as we did in the third line.

³⁹ Too bad.

Here is a lollypop.

⁴⁰ That helps a little.

Let's write the function *eq_main*, which takes two *main* dishes and determines whether they are the same.

⁴¹ Does that mean we need to compare all four possible *main* dishes with each other?

Yes, that is precisely what we mean.

⁴² Here it is.

```
fun eq_main(Steak,Steak)
  = true
| eq_main(Steak,Ravioli)
  = false
| eq_main(Steak,Chicken)
  = false
| eq_main(Steak,Eggplant)
  = false
| eq_main(Ravioli,Steak)
  = false
| eq_main(Ravioli,Ravioli)
  = true
| eq_main(Ravioli,Chicken)
  = false
| eq_main(Ravioli,Eggplant)
  = false
| eq_main(Chicken,Steak)
  = false
| eq_main(Chicken,Ravioli)
  = false
| eq_main(Chicken,Chicken)
  = true
| eq_main(Chicken,Eggplant)
  = false
| eq_main(Eggplant,Steak)
  = false
| eq_main(Eggplant,Ravioli)
  = false
| eq_main(Eggplant,Chicken)
  = false
| eq_main(Eggplant,Eggplant)
  = true
```

Where is its type?

⁴³ Here.

```
eq_main :
  (main * main) → bool
```

How does this type differ from the type of *add_a_steak*?

⁴⁴ It has a star to the left of \rightarrow instead of the right.

Does that mean *eq_main* consumes two things?

⁴⁵ Not really, it consumes a pair of *main* dishes, which we sometimes think of as two dishes.

Here is a shorter version.

```
fun eq_main(Steak,Steak)
  = true
| eq_main(Ravioli,Ravioli)
  = true
| eq_main(Chicken,Chicken)
  = true
| eq_main(Eggplant,Eggplant)
  = true
| eq_main(a_main,another_main)
  = false
```

⁴⁶ This is much shorter than the previous one and it contains far fewer patterns

Yes, once we have defined a function, we may be able to rearrange patterns and make a function shorter.

⁴⁷ That's neat but who could have figured that out?

What is the value of
has_steak(Hummus,Ravioli,Sundae)?

⁴⁸ false.

And
has_steak(Shrimp,Steak,Mousse)?

⁴⁹ true.

Good. What does the function consume?

⁵⁰ A small meal consisting of *meza*, *main*, and *dessert*.

What does it produce?

⁵¹ *bool*.

What is the type of *has_steak*?

⁵² $(meza * main * dessert) \rightarrow bool.$

Could we write the unabridged version of *has_steak*?

⁵³ It would make our fingers too tired.

Let's define just the abridged version of *has_steak*.

⁵⁴ That's easy.

```
fun has_steak(a_meza,Steak,a_dessert)
  = true
  | has_steak(a_meza,a_main,a_dessert)
  = false
```

What is its type?

⁵⁵ It does consume *meza*, a *main* dish, and a *dessert*. So, it seems that this is the type:
 $(meza * main * dessert) \rightarrow bool.$

That's true. But is
 has_steak(5,Steak,true)
nonsense?

⁵⁶ Nearly. If *has_steak* has the type we said it has, then it is nonsense.

So, is it nonsense?

⁵⁷ The definition of *has_steak* does not prevent it from consuming 5 and true.

Then what is the type of this abridged version of *has_steak*?

⁵⁸ We need another Greek letter like α to make the type.

Why?

⁵⁹ Because the first and the third components do not need to belong to the same type. Therefore we must say the third component is arbitrary, yet differs from the first.

Here is the type.

```
has_steak :  
  ( $\alpha$  * main *  $\beta^1$ )  $\rightarrow$  bool
```

•

¹ We use 'b for β , but it is pronounced beta.

⁶⁰ Good, but could we also have written this?

```
has_steak :  
  ( $\beta$  * main *  $\alpha$ )  $\rightarrow$  bool
```

•

Yes, the two types are identical except for the Greek names of the types.

⁶¹ They both say that *has_steak* consumes three things, the first and third belong to arbitrary, distinct types.

Do α and β always stand for different types?

⁶² No, *has_steak* can also consume (5,Ravioli,6).

We won't use any other Greek letters,

⁶³ That's good.

Does it make sense to have *has_steak* consume (5,Ravioli,6)?

⁶⁴ No, *has_steak* should consume only *meza* and *desserts* along with a *main* dish.

Is it possible to restrict the function so that it would consume only good things?

⁶⁵ We could say its type is this.

```
has_steak :  
  (meza * main * dessert)  $\rightarrow$  bool
```

•

Unfortunately, that is only enough for us because we agreed to respect these statements about the types of functions. If we really want to restrict the type of things *has_steak* consumes, we need to combine the bulleted type boxes with the definitions.

⁶⁶ Looks simple. It is obvious where the various underlined pieces come from.

```
fun has_steak(a:meza,Steak,d:dessert):bool  
  = true  
  | has_steak(a:meza,ns,d:dessert):bool  
  = false
```

If it looks simple, why not combine the type of the first version of *add_a_steak* and the second definition to restrict its use, too.

```
fun add_a_steak(x)
  = (x,Steak)
```

```
add_a_steak :
  meza → (meza * main)
```

⁶⁷ Here it is:

```
fun add_a_steak(x:meza):(meza * main)
  = (x,Steak)
```

Relax and enjoy a hot fudge sundae.

⁶⁸ After a delicious Turkish meza platter.

The Fourth Moral

Some functions consume values of star type; some produce values of star type.

5. Couples Are Magnificent, Too



Изображение, защищенное авторским правом

Have we seen this kind of definition before?

¹ What? More pizza!

```
datatype  $\alpha$  pizza =  
  Bottom  
  | Topping of ( $\alpha$  * ( $\alpha$  pizza))
```

Yes, still more pizza, but this one is interesting.

² Yes, we have seen something like this kind of definition before. A type definition using α abbreviates many different type definitions. But isn't this the first **datatype** definition that uses a star?

Yes, it is. Use a **datatype** definition to describe the shape that is like the type *fish pizza* using this definition of *fish*.

```
datatype fish =  
  Anchovy  
  | Lox  
  | Tuna
```

³ Here it is.¹

```
datatype fish pizza =  
  Bottom  
  | Topping of (fish * (fish pizza))
```



¹ Recall that \otimes indicates that this definition is ungrammatical, but this definition expresses the idea best.

Is
 Topping(Anchovy,
 Topping(Tuna,
 Topping(Anchovy,
 Bottom)))
a pizza of type *fish pizza*?

⁴ It is a *fish pizza* provided
 Topping(Tuna,
 Topping(Anchovy,
 Bottom))
is a *fish pizza*, because **Topping** makes these kinds of pizzas.

Is
 Topping(Tuna,
 Topping(Anchovy,
 Bottom))
a *fish pizza*?

⁵ Yes, it too is a *fish pizza*, if
 Topping(Anchovy,
 Bottom)
is a *fish pizza*.

Is
 Topping(Anchovy,
 Bottom)
a *fish pizza*?

⁶ Yes, it is, because Topping constructs a *fish pizza* from Anchovy—a fish—and Bottom—a *fish pizza*.

Is Bottom really a *fish pizza*?

⁷ Yes, because Bottom is at the bottom of many kinds of pizzas. We could also put it at the bottom of an *int pizza*, a *bool pizza*, or a *num pizza*.

What is the value of
 rem_anchovy(
 Topping(Lox,
 Topping(Anchovy,
 Topping(Tuna,
 Topping(Anchovy,
 Bottom))))))?

⁸ It is this *fish pizza*:
 Topping(Lox,
 Topping(Tuna,
 Bottom)).

Is it true that the value of
 rem_anchovy(
 Topping(Lox,
 Topping(Tuna,
 Bottom)))
is
 Topping(Lox,
 Topping(Tuna,
 Bottom))?

⁹ Yes, the pizza that comes out is the same as the one that goes in.

Does rem_anchovy consume *fish pizza* and produce *fish pizza*?

¹⁰ Yes, it does, and it does not consume a *num pizza* or an *int pizza*.

Define *rem_anchovy*. Here is a skeleton.

```
fun rem_anchovy(Bottom)
  = _____
  | rem_anchovy(Topping(Anchovy,p))
  = _____
  | rem_anchovy(Topping(Tuna,p))
  = _____
  | rem_anchovy(Topping(Lox,p))
  = _____
```

```
rem_anchovy :
  (fish pizza) → (fish pizza)
```

¹¹ This is easy by now.

```
fun rem_anchovy(Bottom)
  = Bottom
  | rem_anchovy(Topping(Anchovy,p))
  = rem_anchovy(p)
  | rem_anchovy(Topping(Tuna,p))
  = Topping(Tuna,rem_anchovy(p))
  | rem_anchovy(Topping(Lox,p))
  = Topping(Lox,rem_anchovy(p))
```

Is there a shorter version of *rem_anchovy*?

¹² Yes, we can combine the last two patterns and their answers if we let *t* stand for either Tuna or Lox.

Do we expect you to know that?

¹³ No, but here is the definition.

```
fun rem_anchovy(Bottom)
  = Bottom
  | rem_anchovy(Topping(Anchovy,p))
  = rem_anchovy(p)
  | rem_anchovy(Topping(t,p))
  = Topping(t,rem_anchovy(p))
```

How does
 rem_tuna
differ from
 rem_anchovy?

¹⁴ Not much. It removes Tuna instead of Anchovy. Here is the definition.

```
fun rem_tuna(Bottom)
  = Bottom
  | rem_tuna(Topping(Anchovy,p))
  = Topping(Anchovy,rem_tuna(p))
  | rem_tuna(Topping(Tuna,p))
  = rem_tuna(p)
  | rem_tuna(Topping(Lox,p))
  = Topping(Lox,rem_tuna(p))
```

Where is the type?

¹⁵ Here it is.

```
rem_tuna :  
  (fish pizza) → (fish pizza)
```

•

Can we shorten this definition like we shortened that of *rem_anchovy*?

¹⁶ No, the patterns and answers that are alike are too far apart.

How do the following two definitions of *fish* differ?

```
datatype fish =  
  Anchovy  
| Lox  
| Tuna
```

```
datatype fish =  
  Tuna  
| Lox  
| Anchovy
```

¹⁷ They aren't really different, because they both say that Lox, Anchovy, and Tuna are *fish*. But, if we had chosen the second definition, we would have defined *rem_tuna* like this.

```
fun rem_tuna(Bottom)  
  = Bottom  
| rem_tuna(Topping(Tuna,p))  
  = rem_tuna(p)  
| rem_tuna(Topping(Lox,p))  
  = Topping(Lox,rem_tuna(p))  
| rem_tuna(Topping(Anchovy,p))  
  = Topping(Anchovy,rem_tuna(p))
```

```
rem_tuna :  
  (fish pizza) → (fish pizza)
```

•

Why would we have defined *rem_tuna* like that?

¹⁸ Because we have always ordered the patterns according to the alternatives in the corresponding **datatype** definition.

Can we shorten this new definition of *rem_tuna*?

¹⁹ Yes, because the pair of patterns and answers that are alike are close together.

Do we have to change the definition of *fish* to²⁰ do all that?

No, we don't. The ordering of the patterns does not matter as long as there is one for each alternative in the corresponding **datatype** definition. But we like to keep things in the same order.

Write a shorter version of *rem_tuna*.

```
fun rem_tuna(Bottom)
  = Bottom
| rem_tuna(Topping(Tuna,p))
  = rem_tuna(p)
| rem_tuna(Topping(Lox,p))
  = Topping(Lox,rem_tuna(p))
| rem_tuna(Topping(Anchovy,p))
  = Topping(Anchovy,rem_tuna(p))
```

²¹ Here's one.

```
fun rem_tuna(Bottom)
  = Bottom
| rem_tuna(Topping(Tuna,p))
  = rem_tuna(p)
| rem_tuna(Topping(t,p))
  = Topping(t,rem_tuna(p))
```

Can we combine *rem_anchovy* and *rem_tuna* into one function?

²² Yes, but when we use the combined function, we need to say which kind of fish we want to remove.

What is a good name for the combined function?

²³ How about *rem_fish*?

How do we use *rem_fish*?

²⁴ We give it a pair of things. The first component could be the kind of fish we want to remove and the second one could be the pizza.

Could we also give it a pair where the second component is the kind of fish we want to remove and the first one is the pizza?

²⁵ Yes, it doesn't matter as long as we stick to one choice.

What would be the type of *rem_fish* if we chose the second alternative?

²⁶ That's easy:
 $((\text{fish pizza}) * \text{fish}) \rightarrow (\text{fish pizza}).$

But, let's use the first one.

²⁷ Then *rem_fish* consumes a pair that consists of a *fish* and a *fish pizza*.

Here is the definition of *rem_fish*.

²⁸ As we will see, it could have been worse.

```
fun rem_fish(x, Bottom)
  = Bottom
| rem_fish(Tuna, Topping(Tuna, p))
  = rem_fish(Tuna, p)
| rem_fish(Tuna, Topping(t, p))
  = Topping(t, rem_fish(Tuna, p))
| rem_fish(Anchovy, Topping(Anchovy, p))
  = rem_fish(Anchovy, p)
| rem_fish(Anchovy, Topping(t, p))
  = Topping(t, rem_fish(Anchovy, p))
| rem_fish(Lox, Topping(Lox, p))
  = rem_fish(Lox, p)
| rem_fish(Lox, Topping(t, p))
  = Topping(t, rem_fish(Lox, p))
```

```
rem_fish :
  (fish * (fish pizza)) → (fish pizza) •
```

Isn't this clumsy?

Describe in your words how it could have been worse.

²⁹ Here are ours:
"The pattern

rem_fish(Tuna, Topping(*t*, *p*))

matches all pairs that consist of Tuna and a *fish pizza* whose topping is not Tuna. For the long version of *rem_fish* we would have used two different patterns:

rem_fish(Tuna, Topping(Anchovy, *p*))

and

rem_fish(Tuna, Topping(Lox, *p*)).

And, we would also have needed an answer for each pattern."

Write the unabridged version of *rem_fish*.

³⁰ It has three more patterns than the short one.

```
fun rem_fish(x, Bottom)
  = Bottom
| rem_fish(Tuna, Topping(Tuna, p))
  = rem_fish(Tuna, p)
| rem_fish(Tuna, Topping(Anchovy, p))
  = Topping(Anchovy, rem_fish(Tuna, p))
| rem_fish(Tuna, Topping(Lox, p))
  = Topping(Lox, rem_fish(Tuna, p))
| rem_fish(Anchovy, Topping(Anchovy, p))
  = rem_fish(Anchovy, p)
| rem_fish(Anchovy, Topping(Lox, p))
  = Topping(Lox, rem_fish(Anchovy, p))
| rem_fish(Anchovy, Topping(Tuna, p))
  = Topping(Tuna, rem_fish(Anchovy, p))
| rem_fish(Lox, Topping(Lox, p))
  = rem_fish(Lox, p)
| rem_fish(Lox, Topping(Anchovy, p))
  = Topping(Anchovy, rem_fish(Lox, p))
| rem_fish(Lox, Topping(Tuna, p))
  = Topping(Tuna, rem_fish(Lox, p))
```

If we add another kind of fish to our **datatype**, what happens to the short function?

³¹ We have to add two patterns and two answers.

If we add another kind of fish to our **datatype**, what happens to the unabridged version?

³² We have to add one pattern and one answer for each old kind of fish and four patterns and answers for the new kind.

Why does the unabridged version get so large?

³³ Because we must compare each kind of fish to every other kind of fish, including itself. And the first pattern is always a test for **Bottom**.

Does that mean the unabridged version for five fish contains 26 patterns?

³⁴ Yes, and for six fish it would be 37. Worse, if n is the number of fish in a **datatype**, the number of patterns needed for the unabridged version is $n^2 + 1$.

Is there a shorter way to determine whether two fish are the same?

³⁵ Could we use the same name in one pattern twice?

```
fun rem_fish(x, Bottom)
  = Bottom
| rem_fish(x, Topping(x, p))
  = rem_fish(x, p)
| rem_fish(x, Topping(t, p))
  = Topping(t, rem_fish(x, p))
```



Wouldn't that be great? Unfortunately, using the same name *twice* in a pattern is ungrammatical.

³⁶ Sigh.

Let's define the function *eq_fish*, which determines whether two given *fish* are equal.

³⁷ That function consumes a pair of *fish* and produces a *bool*.

The unabridged version of *eq_fish* is huge.

³⁸ It is only four lines long.

```
fun eq_fish(Anchovy, Anchovy)
  = true
| eq_fish(Anchovy, Lox)
  = false
| eq_fish(Anchovy, Tuna)
  = false
| eq_fish(Lox, Anchovy)
  = false
| eq_fish(Lox, Lox)
  = true
| eq_fish(Lox, Tuna)
  = false
| eq_fish(Tuna, Anchovy)
  = false
| eq_fish(Tuna, Lox)
  = false
| eq_fish(Tuna, Tuna)
  = true
```

```
fun eq_fish(Anchovy, Anchovy)
  = true
| eq_fish(Lox, Lox)
  = true
| eq_fish(Tuna, Tuna)
  = true
| eq_fish(a_fish, another_fish)
  = false
```

```
eq_fish :
  (fish * fish) → bool
```



Write the abridged version and provide a type?

What is the value of
`eq_fish(Anchovy, Anchovy)`?

³⁹ It is `true`, unlike `eq_fish(Anchovy, Tuna)`.

Here is the shortest version of `rem_fish` yet.

```
fun rem_fish(x, Bottom)
  = Bottom
  | rem_fish(x, Topping(t, p))
  = if eq_fish(t, x)
    then rem_fish(x, p)
    else Topping(t, (rem_fish(x, p)))
```

⁴⁰ Yes, it contains

```
if exp1
  then exp2
  else exp3,
```

which we haven't seen before. How do we determine its type?

Is there anything new?

To determine its type, we first make sure that the type of `exp1` is `bool`, and then we determine the types of `exp2` and `exp3`.

⁴¹ And these two need to be the same because the value of either one can be the result of the entire expression. Correct?

Yes, great guess. Does this version of `rem_fish` still have the type
 $(\text{fish} * (\text{fish pizza})) \rightarrow (\text{fish pizza})$?

⁴² Yes, since both `rem_fish` and `Topping` produce `fish pizza`, `rem_fish` produces `fish pizza`, no matter which of `exp2` or `exp3` is evaluated.

How does that new version differ from this ungrammatical one?

```
fun rem_fish(x, Bottom)
  = Bottom
  | rem_fish(x, Topping(x, p))
  = rem_fish(x, p)
  | rem_fish(x, Topping(t, p))
  = Topping(t, rem_fish(x, p))
```



⁴³ Not too much. The shortest version uses `eq_fish` to compare the two kinds of fish; this one uses an ungrammatical pattern.

Let's try it out with the shortest version:
`rem_fish(Anchovy,
 Topping(Anchovy,
 Bottom)).`

⁴⁴ It does not match the first pattern, because the pizza is not `Bottom`.

Does the second pattern match?

⁴⁵ If x is Anchovy, t is Anchovy, and p is Bottom, then it matches.

What next?

⁴⁶ Next we need to compare t with x , which are equal, so $eq_fish(t, x)$ is true.

Therefore, we take $rem_fish(x, p)$ as the answer.

⁴⁷ Since p is Bottom, the result of that expression is Bottom, and that is also the result of
 $rem_fish(Anchovy,$
 $Topping(Anchovy,$
 $Bottom))$.

What is the value of
 $rem_fish(Tuna,$
 $Topping(Anchovy,$
 $Topping(Tuna,$
 $Topping(Anchovy,$
 $Bottom))))$?

⁴⁸ Again, the first pattern doesn't match, but the other one does, if x is Tuna, t is Anchovy, and p is
 $Topping(Tuna,$
 $Topping(Anchovy,$
 $Bottom))$.

What is $eq_fish(t, x)$ if t is Anchovy and x is Tuna?

⁴⁹ false.

So what is the answer?

⁵⁰ The answer is
 $Topping(Anchovy,$
 $rem_fish(Tuna,$
 $Topping(Tuna,$
 $Topping(Anchovy,$
 $Bottom))))$,
which is what follows the pattern and the = sign with x replaced by Tuna, t replaced by Anchovy, and p replaced by
 $Topping(Tuna,$
 $Topping(Anchovy,$
 $Bottom))$.

Which pattern does

```
rem_fish(Tuna,  
  Topping(Tuna,  
    Topping(Anchovy,  
      Bottom)))
```

match?

⁵¹ It matches the second one again if x is Tuna, t is Tuna, and p is
 Topping(Anchovy,
 Bottom).

And how do we continue?

⁵² We determine the value of
 rem_fish(Tuna,
 Topping(Anchovy,
 Bottom)),
because we want to remove Tuna.

Is
 Topping(Anchovy,
 Bottom)

the value of
 rem_fish(Tuna,
 Topping(Anchovy,
 Bottom))?

⁵³ Yes, because the pizza does not contain any
 Tuna.

So what is the final answer?

⁵⁴ We still need to top it with anchovy:
 Topping(Anchovy,
 Topping(Anchovy,
 Bottom)).

Does

```
rem_int(3,  
  Topping(2,  
    Topping(3,  
      Topping(2,  
        Bottom))))
```

look familiar?

⁵⁵ Yes, it looks like what we just evaluated.

What does *rem_int* do?

⁵⁶ It removes *ints* from *int pizzas* just as
 rem_fish removes *fish* from *fish pizzas*.

With *eq_int*,¹ define *rem_int*.

¹ You must define *eq_int* as
`fun eq_int(n:int,m:int) = (n = m).`

⁵⁷ That's easy, it is nearly identical to the definition of *rem_fish*.

```
fun rem_int(x,Bottom)
  = Bottom
  | rem_int(x,Topping(t,p))
  = if eq_int(t,x)
    then rem_int(x,p)
    else Topping(t,(rem_int(x,p)))
```

```
rem_int :
  (int * (int pizza)) → (int pizza)
```

Describe how *rem_fish* differs from *rem_int*.

⁵⁸ Here is what is on our mind:
“They look alike, but they differ in the types of the things that they consume and produce, and therefore in how they compare toppings.”

Can we define one function that removes toppings from many kinds of pizza?

⁵⁹ Yes, but not until chapter 8.

What is the value of
subst_fish(Lox,Anchovy,
 Topping(Anchovy,
 Topping(Tuna,
 Topping(Anchovy,
 Bottom))))?

⁶⁰ It is the same pizza with all instances of Anchovy replaced by Lox:
 Topping(Lox,
 Topping(Tuna,
 Topping(Lox,
 Bottom))).

What value does *subst_fish* consume?

⁶¹ It consumes a triple whose first two components are of type *fish* and whose last component is a *fish pizza*.

And what does it produce?

⁶² It always produces a *fish pizza*.

What is the value of

```
subst_int(5,3,  
  Topping(3,  
    Topping(2,  
      Topping(3,  
        Bottom))))?)
```

⁶³ It is the same pizza with all 3s replaced by 5s:

```
Topping(5,  
  Topping(2,  
    Topping(5,  
      Bottom)))).
```

What value does *subst_int* consume?

⁶⁴ It consumes a triple whose first two components are of type *int* and whose last component is an *int pizza*.

And what does it produce?

⁶⁵ It always produces an *int pizza*.

We can define *subst_fish*.

```
fun subst_fish(n,a,Bottom)  
  = Bottom  
  | subst_fish(n,a,Topping(t,p))  
  = if eq_fish(t,a)  
    then Topping(n,subst_fish(n,a,p))  
    else Topping(t,subst_fish(n,a,p))
```

```
subst_fish :  
(fish * fish * (fish pizza)) → (fish pizza) •
```

⁶⁶ To get from *subst_fish* to *subst_int*, we just need to substitute *fish* by *int* everywhere.

```
fun subst_int(n,a,Bottom)  
  = Bottom  
  | subst_int(n,a,Topping(t,p))  
  = if eq_int(t,a)  
    then Topping(n,subst_int(n,a,p))  
    else Topping(t,subst_int(n,a,p))
```

```
subst_int :  
(int * int * (int pizza)) → (int pizza) •
```

Can we define *subst_int*?

eq_int(17,0)?

⁶⁷ false,
because 17 and 0 are different.

eq_int(17,Tuna)?

⁶⁸ This is nonsense,¹ because 17 and Tuna belong to two different types.

¹ Remember that we use the word "nonsense" to refer to expressions that have no type.

What is the value of

```
eq_num(  
  One_more_than(  
    Zero),  
  One_more_than(  
    Zero))?
```

⁶⁹ true,

because both values are constructed with `One_more_than` and the same component.

Define `eq_num`, but don't forget that it takes two values.

⁷⁰ It is easy to write the unabridged version if we use two patterns for each value that it consumes.

```
fun eq_num(Zero,Zero)  
  = true  
| eq_num(One_more_than(n),Zero)  
  = false  
| eq_num(Zero,One_more_than(m))  
  = false  
| eq_num(  
  One_more_than(n),  
  One_more_than(m))  
  = eq_num(n,m)
```

Define the abridged version. Here is a version where we reordered some patterns. Can the last two be combined?

⁷¹ No problem.

```
fun eq_num(Zero,Zero)  
  = true  
| eq_num(  
  One_more_than(n),  
  One_more_than(m))  
  = eq_num(n,m)  
| eq_num(One_more_than(n),Zero)  
  = false  
| eq_num(Zero,One_more_than(m))  
  = false
```

```
fun eq_num(Zero,Zero)  
  = true  
| eq_num(  
  One_more_than(n),  
  One_more_than(m))  
  = eq_num(n,m)  
| eq_num(n,m)  
  = false
```

No problem?

⁷² Not if we start from a correct program and carefully transform it, step by step.

Perhaps it is time to digest something
besides this book.

⁷³ Great idea. How about a granola bar and a
walk?

The Fifth Moral

*Write the first draft of a function fol-
lowing all the morals. When it is cor-
rect and no sooner, simplify.*

6. Oh My, It's Full of Stars!



Изображение, защищенное авторским правом

Is ¹ Yes.
Flat(Apple,
Flat(Peach,
Bud))
a flat *tree*?

Is ² Yes, it is also a flat *tree*.
Flat(Pear,
Bud)
a flat *tree*?

And how about ³ No, it contains Split, so it can't be flat.
Split(
Bud,
Flat(Fig,
Split(
Bud,
Bud))))?

Here is one more example: ⁴ No, it isn't flat either.
Split(
Split(
Bud,
Flat(Lemon,
Bud)),
Flat(Fig,
Split(
Bud,
Bud))).
Is it flat?

Ready to go? ⁵ Sure. Let's define the **datatypes** we need to
make this work.

Here are some fruits.

```
datatype fruit =  
  Peach  
| Apple  
| Pear  
| Lemon  
| Fig
```

Let's say all *trees* are either flat, split, or bud. Formulate the **datatype** for *trees*.

⁶ It does not differ too much from the **datatypes** we have seen before.

```
datatype tree =  
  Bud  
| Flat of fruit * tree  
| Split of tree * tree
```

How is it different from all the other **datatypes** we have seen before?

⁷ The name of the new **datatype** occurs twice in one (the last) alternative.

How many patterns does the definition of *flat_only* contain?

⁸ Three, because it consumes *trees*, and the **datatype** *tree* contains three alternatives.

What type of value does *flat_only* produce?

⁹ *bool*.

What function does *flat_only* remind us of?

¹⁰ *only_onions*.

Here is a skeleton for *flat_only*.

```
fun flat_only(Bud)  
  = _____  
| flat_only(Flat(f,t))  
  = _____  
| flat_only(Split(s,t))  
  = _____
```

Fill in the blanks and supply the type.

¹¹ That's easy now.

```
fun flat_only(Bud)  
  = true  
| flat_only(Flat(f,t))  
  = flat_only(t)  
| flat_only(Split(s,t))  
  = false
```

```
flat_only :  
tree → bool
```

Define the function *split_only*, which checks whether a tree is constructed with **Split** and **Bud** only.

¹² Here is the easy part.

```
fun split_only(Bud)
  = true
| split_only(Flat(f,t))
  = false
| split_only(Split(s,t))
  = _____
```

What is difficult about the last line?

¹³ We need to check whether both *s* and *t* are split trees.

Isn't that easy?

¹⁴ Yes, we just use *split_only* on *s* and *t*.

And then?

¹⁵ Then we need to know that both are true.

Doesn't that mean we need to know that *split_only(t)* is true if *split_only(s)* is true?

¹⁶ Yes.

Do we need to know whether *split_only(t)* is true if *split_only(s)* is false?

¹⁷ No, then the answer is false.

Finish the definition of *split_only* using

```
if ...
then ...
else ... .
```

¹⁸ Now we can do it.

```
fun split_only(Bud)
  = true
| split_only(Flat(f,t))
  = false
| split_only(Split(s,t))
  = if1 split_only(s)
      then split_only(t)
      else false
```

¹ We could have written this if-expression as *split_only(s) andalso split_only(t)*.

```
split_only :
  tree → bool
```

Give an example of a *tree* for which *split_only*¹⁹ responds with true. There is a trivial one: **Bud**.

How about one with five uses of **Split**?

²⁰ Here is one:

```
Split(
  Split(
    Bud,
    Split(
      Bud,
      Bud)),
  Split(
    Bud,
    Split(
      Bud,
      Bud))).
```

Does this *tree* contain any fruit?

²¹ No tree for which *split_only* is true contains any fruit.

Here is one version of the definition of the function *contains_fruit*.

```
fun contains_fruit(Bud)
  = false
| contains_fruit(Flat(f,t))
  = true
| contains_fruit(Split(s,t))
  = if1 contains_fruit(s)
    then true
    else contains_fruit(t)
```

```
contains_fruit :
  tree → bool
```

•

²² We can use *split_only*, which already checks whether a *tree* contains a **Flat**.

```
fun contains_fruit(x)
  = if1 split_only(x)
    then false
    else true
```

Write a shorter one.

¹ We could have written this **if**-expression as `contains_fruit(s) orelse contains_fruit(t)`.

¹ We could have written this **if**-expression as `not(split_only(x))`.

What is the height of

²³ 3.

```
Split(  
  Split(  
    Bud,  
    Flat(Lemon,  
          Bud)),  
  Flat(Fig,  
        Split(  
          Bud,  
          Bud))))?
```

What is the height of

²⁴ 2.

```
Split(  
  Bud,  
  Flat(Lemon,  
        Bud))?
```

What is the height of

²⁵ 1.

```
Flat(Lemon,  
      Bud)?
```

What is the height of

²⁶ 0.

```
Bud?
```

So what is the height of a *tree*?

²⁷ The height of a tree is the distance from the root to the highest bud in the tree.

Does *height* consume a *tree*?

²⁸ Yes, and it produces an *int*.

What is the value of

²⁹ 3, isn't it?

```
height(  
  Flat(Fig,  
        Flat(Lemon,  
              Flat(Apple,  
                    Bud))))?
```

What is the value of

```
height(  
  Split(  
    Split(  
      Bud,  
      Bud),  
    Flat(Fig,  
      Flat(Lemon,  
        Flat(Apple,  
          Bud))))))?
```

³⁰ 4.

Why is the height 4?

³¹ Because the value of

```
height(  
  Split(  
    Bud,  
    Bud))
```

is 1, the value of

```
height(  
  Flat(Fig,  
    Flat(Lemon,  
      Flat(Apple,  
        Bud))))
```

is 3, and the larger of the two numbers is 3.

And how do we get from 3 to 4?

³² We need to add 1 to the larger of the numbers so that we don't forget the `Split` at the root of the tree.

Define the function *larger_of*.

³³ What does it consume?

It consumes a pair of *ints* and produces an *int*.

³⁴ Well, then it must be this.

```
larger_of :  
  (int * int) → int
```

•

```
fun larger_of(n,m)  
  = if less_than1(n,m)  
    then m  
    else n
```

¹ You must define `less_than` as
`fun less_than(n:int,m:int) = (n < m).`

Here is *height*.

```
fun height(Bud)
  = 0
| height(Flat(f,t))
  = 1 + height(t)
| height(Split(s,t))
  = 1 + larger_of(height(s),height(t))
```

³⁵ And here is its type.

```
height :
  tree → int
```

What is the value of
height(Split(Bud,Bud))?

³⁶ 1, of course.

And why is it 1?

³⁷ Because *height*(Bud) is 0 and the larger of 0 and 0 is 0. And one more than 0 is 1.

What is the value of
subst_in_tree(Apple, Fig,
Split(
Split(
Flat(Fig,
Bud),
Flat(Fig,
Bud)),
Flat(Fig,
Flat(Lemon,
Flat(Apple,
Bud)))))?

³⁸ That's also easy. We replace all Figs by Apples:
Split(
Split(
Flat(Apple,
Bud),
Flat(Apple,
Bud)),
Flat(Apple,
Flat(Lemon,
Flat(Apple,
Bud))))).

Do we need to define *eq_fruit* before we define *subst_in_tree*? Here is its type.

```
eq_fruit :  
  (fruit * fruit) → bool
```

³⁹ How could you know, but we do need it!

```
fun eq_fruit(Peach,Peach)  
  = true  
| eq_fruit(Apple,Apple)  
  = true  
| eq_fruit(Pear,Pear)  
  = true  
| eq_fruit(Lemon,Lemon)  
  = true  
| eq_fruit(Fig,Fig)  
  = true  
| eq_fruit(a_fruit,another_fruit)  
  = false
```

How many lines would *eq_fruit* be if we had twenty-five different fruits?

⁴⁰ When you have counted them all, you can have some apple juice.

Define the function *subst_in_tree*.

⁴¹ It's like *subst_fish* and *subst_int* from the end of chapter 5.

```
fun subst_in_tree(n,a,Bud)  
  = Bud  
| subst_in_tree(n,a,Flat(f,t))  
  = if eq_fruit(f,a)  
    then Flat(n,subst_in_tree(n,a,t))  
    else Flat(f,subst_in_tree(n,a,t))  
| subst_in_tree(n,a,Split(s,t))  
  = Split(  
    subst_in_tree(n,a,s),  
    subst_in_tree(n,a,t))
```

```
subst_in_tree :  
  (fruit * fruit * tree) → tree
```

How many times does Fig occur in

⁴² 3.

```
Split(
  Split(
    Flat(Fig,
      Bud),
    Flat(Fig,
      Bud)),
  Flat(Fig,
    Flat(Lemon,
      Flat(Apple,
        Bud)))))?
```

Write the function *occurs*.

⁴³ This is so easy; just follow the patterns.

```
fun occurs(a,Bud)
  = 0
| occurs(a,Flat(f,t))
  = if eq_fruit(f,a)
    then 1 + occurs(a,t)
    else occurs(a,t)
| occurs(a,Split(s,t))
  = occurs(a,s) + occurs(a,t)
```

```
occurs :
  (fruit * tree) → int
```

Do you like your fruit with yogurt?

⁴⁴ We prefer coconut sorbet.

Is it true that
 An_atom(5)
is an *sexp*?

⁴⁵ Yes,
 because An_atom is one of the two
 constructors of *int sexp*.

Is it true that
 An_atom(Fig)
is an *sexp*?

⁴⁶ Yes,
 because An_atom is one of the two
 constructors of *fruit sexp*.

Is it true that
 A_slist(Empty)
is an *sexp*?

⁴⁷ Yes,
 because A_slist is the other constructor of
 int sexp.

Is it also true that
 A_slist(Empty)
is an *sexp*?

⁴⁸ Yes,
 because A_slist is the other constructor of
 fruit sexp.

Is it true that
 Scons(An_atom(5),
 Scons(An_atom(13),
 Scons(An_atom(1),
 Empty)))
is an *int slist*?

⁴⁹ Yes,
 because here Scons constructs *int slists*
 from *int sexps* and *int slists*.

Is it also true that
 Scons(An_atom(Fig),
 Empty)
is a *fruit slist*?

⁵⁰ Yes,
 because Scons also constructs *fruit slist*
 from *fruit sexps* and *fruit slists*.

Okay, so here are two new shapes.

⁵¹ α slist and α sexp.

```
datatype
   $\alpha$  slist =
    Empty
  | Scons of (( $\alpha$  sexp) * ( $\alpha$  slist))
and
   $\alpha$  sexp =
    An_atom of  $\alpha$ 
  | A_slist of ( $\alpha$  slist)
```

What are the two shapes?

Why are the two definitions separated by
and?

⁵² The first definition, α slist, refers to the
second, α sexp; and the second refers to the
first.

Do such mutually self-referential **datatypes** ⁵³ Always.
lead to mutually self-referential functions?

How many times does Fig occur in ⁵⁴ Twice.
Scons(An_atom(Fig),
Scons(An_atom(Fig),
Scons(An_atom(Lemon),
Empty)))?

What is the value of ⁵⁵ 2, again.
occurs_in_slist(Fig,
Scons(A_slist(
Scons(An_atom(Fig),
Scons(An_atom(Peach),
Empty))),
Scons(An_atom(Fig),
Scons(An_atom(Lemon),
Empty))))?

And what does ⁵⁶ 1.
occurs_in_sexp(Fig,
A_slist(
Scons(An_atom(Fig),
Scons(An_atom(Peach),
Empty))))
evaluate to?

Here are the skeletons of *occurs_in_slist* and *occurs_in_sexp*.

```

fun
  occurs_in_slist(a,Empty)
    = _____
  | occurs_in_slist(a,Scons(s,y))
    = _____
and
  occurs_in_sexp(a,An_atom(b))
    = if eq_fruit(b,a)
      then 1
      else 0
  | occurs_in_sexp(a,A_slist(y))
    = _____

```

```

occurs_in_slist :
  (fruit * fruit slist) → int

```

Fill in the blanks. Also provide the type for *occurs_in_sexp*.

⁵⁷ The blanks are easy now, because they just stand for the obvious answers.

```

fun
  occurs_in_slist(a,Empty)
    = 0
  | occurs_in_slist(a,Scons(s,y))
    = occurs_in_sexp(a,s) +
      occurs_in_slist(a,y)
and
  occurs_in_sexp(a,An_atom(b))
    = if eq_fruit(b,a)
      then 1
      else 0
  | occurs_in_sexp(a,A_slist(y))
    = occurs_in_slist(a,y)

```

```

occurs_in_sexp :
  (fruit * fruit sexp) → int

```

Define *subst_in_slist* and *subst_in_sexp*. Here are their types.

```

subst_in_slist :
  (fruit * fruit * fruit slist) → fruit slist

```

```

subst_in_sexp :
  (fruit * fruit * fruit sexp) → fruit sexp

```

⁵⁸ That is no problem either.

```

fun
  subst_in_slist(n,a,Empty)
    = Empty
  | subst_in_slist(n,a,Scons(s,y))
    = Scons(
      subst_in_sexp(n,a,s),
      subst_in_slist(n,a,y))
and
  subst_in_sexp(n,a,An_atom(b))
    = if eq_fruit(b,a)
      then An_atom(n)
      else An_atom(b)
  | subst_in_sexp(n,a,A_slist(y))
    = A_slist(subst_in_slist(n,a,y))

```

Does that mean we should check in *rem_from_slist* whether the *sexp* inside of *Scons* is an atom?

⁶³ Yes, we should check that and whether the atom is the one that is to be removed.

Here are the refined skeletons.

```
fun
  rem_from_slist(a, Empty)
  = Empty
| rem_from_slist(a, Scons(s, y))
  = if eq_fruit_in_atom(a, s)
      then rem_from_slist(a, y)
      else Scons(
          rem_from_sexp(a, s),
          rem_from_slist(a, y))
and
  rem_from_sexp(a, An_atom(b))
  = _____
| rem_from_sexp(a, A_slist(y))
  = A_slist(rem_from_slist(a, y))
```

⁶⁴ We cannot know because we have never seen *eq_fruit_in_atom* before.

Is *rem_from_sexp* ever applied to a fruit and an atom constructed from the same fruit?

Voilà.

```
fun eq_fruit_in_atom(a, An_atom(s))
  = eq_fruit(a, s)
| eq_fruit_in_atom(a, fruit, A_slist(y))
  = false
```

⁶⁵ That's not difficult.

```
eq_fruit_in_atom :
  (fruit * fruit sexp) → bool
```

What is the type of *eq_fruit_in_atom*?

And what does it do?

⁶⁶ It consumes a *fruit* and a *fruit sexp* and determines whether the latter is an atom constructed from the given *fruit*.

Is *rem_from_sexp* ever applied to a fruit and an atom constructed from the same fruit?

⁶⁷ Not in *rem_from_slist*, because *rem_from_sexp* is only applied when *eq_fruit_in_atom(a, s)* is false.

What is the answer to the first pattern in *rem_from_sexp*?

⁶⁸ Since it is never applied to two identical atoms, the answer is always *An_atom(b)*. Hence, these are the complete mutually self-referential definitions.

```
fun
  rem_from_slist(a,Empty)
  = Empty
| rem_from_slist(a,Scons(s,y))
  = if eq_fruit_in_atom(a,s)
      then rem_from_slist(a,y)
      else Scons(
                rem_from_sexp(a,s),
                rem_from_slist(a,y))
and
  rem_from_sexp(a,An_atom(b))
  = An_atom(b)
| rem_from_sexp(a,A_slist(y))
  = A_slist(rem_from_slist(a,y))
```

Here are two skeletons that are similar to the first two.

```
fun
  rem_from_slist(a,Empty)
  = Empty
| rem_from_slist(a,Scons(An_atom(b),y))
  = _____
and
  rem_from_sexp(a,An_atom(b))
  = _____
| rem_from_sexp(a,A_slist(y))
  = A_slist(rem_from_slist(a,y))
```

⁶⁹ The only change is in the second pattern of *rem_from_slist*. The new pattern says that the first item of the slist must be an atom.

What changed?

What is the answer that corresponds to that pattern?

⁷⁰ The answer depends on *a* and *b*. If they are the same, it is

rem_from_slist(a,y)

otherwise, it is

Scons(An_atom(b),rem_from_slist(a,y)).

Can *rem_from_slist* match all possible α *slists*?

⁷¹ No, not if the first element is an α *sexp* that was constructed by *A_slist*.

Let's add another pattern to the skeletons.

⁷² Something like this.

```
fun
  rem_from_slist(a, Empty)
  = Empty
| rem_from_slist(a, Scons(An_atom(b), y))
  = if eq_fruit(a, b)
      then rem_from_slist(a, y)
      else Scons(
                An_atom(b),
                rem_from_slist(a, y))
| rem_from_slist(a, Scons(A_slist(x), y))
  = _____
and
  rem_from_sexp(a, An_atom(b))
  = _____
| rem_from_sexp(a, A_slist(y))
  = A_slist(rem_from_slist(a, y))
```

What is the answer for the last pattern in *rem_from_slist*?

⁷³ We need to remove all *a*'s from the *slist* *x* and from the *slist* *y*.

Does that mean we can use

rem_from_slist(*a*, *x*)

and

rem_from_slist(*a*, *y*)?

⁷⁴ Yes.

And what do we do with the results?

⁷⁵ We *Scons* them back together again.

What is the type of this function?

```
fun identity(x)  
  = x
```

¹ Whatever it consumes is what it produces.

```
identity :  
   $\alpha \rightarrow \alpha$ 
```

What does $\alpha \rightarrow \alpha$ mean?

² It means that *identity* is a function that consumes and produces values of the same type, no matter what the type is.

What does “no matter what the type is” mean?

³ Here are our words:
“Pick an arbitrary type. Then, *identity* consumes and produces values of the chosen type.”

And what does the word “arbitrary” mean?

⁴ Our words again.
“It means that there is no relationship between the type that you choose and the type that we choose.”

What is the type of *true_maker*?

```
fun true_maker(x)  
  = true
```

⁵ It always produces **true**.

```
true_maker :  
   $\alpha \rightarrow \text{bool}$ 
```

Was that easy?

⁶ Of course, *true_maker* consumes values of any type and always produces a *bool*.

Make up a value of the type *bool_or_int*.

```
datatype bool_or_int =  
  Hot of bool  
  | Cold of int
```

⁷ Here is one: **Hot(true)**.

What is the type of `Hot(true)`?

⁸ `bool_or_int`.

And how about another value of this type?

⁹ `Cold(10)`.

What is the type of `Cold(5)`?

¹⁰ `bool_or_int`.

What is the type of `hot_maker`?

```
fun hot_maker(x)
  = Hot
```

¹¹ It must also start with α , because it can consume anything.

And what does it produce?

¹² It produces `Hot`.

What is the type of `Hot(true)`?

¹³ `bool_or_int`, as we mentioned earlier.

What is the type of `true`?

¹⁴ `bool`.

So `Hot` is of type ...

¹⁵ ... `bool \rightarrow bool_or_int`.

Does that mean `Hot` is a function?

¹⁶ Yes, absolutely.

Did we just agree that constructors are functions?

¹⁷ Those constructors that are followed by **of** in the **datatype** definition are indeed functions.

Then what is the type of `hot_maker`?

¹⁸ It must be this.

```
hot_maker :
   $\alpha \rightarrow (bool \rightarrow bool\_or\_int)$ 
```

Does that mean *hot_maker* is a function?

¹⁹ Yes, we defined it that way.

Here is *help*, a new function definition.

```
fun help(f)
  = Hot(
    true_maker(
      if true_maker(_____)
      then f
      else true_maker))
```

```
help :
  (α → bool) → bool_or_int
```

²⁰ No,
because *true_maker* consumes all types of values, *e.g.*, *true*, *6*, *Hot*, and so on.

```
fun help(f)
  = Hot(
    true_maker(
      if true_maker(5)
      then f
      else true_maker))
```

Does it matter whether the blank is replaced by *true* or *5*?

What is the difference between

$\alpha \rightarrow (bool \rightarrow bool_or_int)$

and

$(\alpha \rightarrow bool) \rightarrow bool_or_int$?

²¹ The difference is the placement of the matching parentheses. In the first type, the parentheses enclose the last two types, *bool* and *bool_or_int*, and in the second type the parentheses enclose the first two types, α and *bool*.

Are they really different?

²² Yes, one consumes a function and the other produces one.

Does that mean functions can consume functions?

²³ Yes, and, as we have already seen, they can produce them, too.

Does that mean functions are values?

²⁴ Yes, functions are values, too.

How do we determine the type of the values that *help* produces?

²⁵ That's easy. We know that *Hot* always returns a *bool_or_int*, which means that *help* must be of type
_____ $\rightarrow bool_or_int$.

How do we determine the type of the values that *help* consumes?

²⁶ That's tricky.

What is the type of the values that *Hot* consumes?

²⁷ *bool*.

Does *true_maker* produce a *bool*?

²⁸ Yes, it does. We said so earlier.

Is it important that *Hot* consumes *bools* and that *true_maker* produces them?

²⁹ Yes, because whatever *true_maker* produces is consumed by *Hot* in the definition of *help*.

What is the type of the values that *true_maker* consumes?

³⁰ It consumes values of any type and therefore it doesn't matter how we fill in the blank.

What is the type of
 if *true_maker*(_____)
 then *f*
 else *true_maker*?

³¹ It doesn't matter, because the result of this expression is consumed by *true_maker* and *true_maker* consumes values of any type.

Does it matter that
 if *true_maker*(_____)
 then *f*
 else *true_maker*
has a type?

³² Although it doesn't matter which type it has, it matters a lot that it has a type. If the expression didn't have a type, it would be nonsense.

How do we determine the type of
 if *exp*₁
 then *exp*₂
 else *exp*₃?

³³ It is the type of *exp*₂ or the type of *exp*₃, because their types must be the same.

What is the type of *true_maker*?

³⁴ $\alpha \rightarrow \text{bool}$.

What does that mean?

⁴³ Every value of type *chain* is a pair. The first component of the pair must be an *int*. That part is easy. But, the second component must be a function.

Haven't we just seen that functions are values?

⁴⁴ Yes, but the kind of function we need now is strange.

What is strange about it?

⁴⁵ It consumes an *int*, which is also easy, but it produces a *chain*, which brings us back to the original problem.

Does that mean the function's type is $int \rightarrow chain$?

⁴⁶ That is what the **datatype** requires.

Here is a start at such a function.

```
fun ints(n)
  = Link(n + 1, _____)
```

⁴⁷ It clearly consumes an *int* and the answer it constructs is a *chain*.

```
ints :
  int → chain
```

•

What is the type of *ints*?

How must we fill in the blank?

⁴⁸ As we said before, the blank must be filled in with a function of type $int \rightarrow chain$.

Don't we have such a function?

⁴⁹ Only one: *ints*.

Fill in the blank now.

⁵⁰ Now it is easy.

```
fun ints(n)
  = Link(n + 1, ints)
```

What is *some_ints*(0)?

⁷⁴ It is a *chain* of all those numbers (larger than 1) that are evenly divisible by 5 or 7.

How many patterns do we need for a function that consumes *chains*?

⁷⁵ One,
because there is only one alternative in the **datatype** definition of *chain*.

Fill in the blanks in this skeleton.

```
fun chain_item(n, Link(i, f))  
  = if eq_int(n, 1)  
    then _____  
    else chain_item(n - 1, _____)
```

⁷⁶ The first blank is easy. The result must be an *int*, so it can only be *i* or *n*. Since we know that *n* is 1, we pick *i*.

```
chain_item :  
(int * chain) → int
```

•

Why is *i* a good answer and not just an answer of the right type?

⁷⁷ We are looking for the *n*th *int* in the *chain*. Since *i* is the first (i.e., 1st) element of the *chain*, the result is *i* when *n* is 1.

Could the answer have been 17?

⁷⁸ The type is right, but who would want to define a function that always returns 17?

What is the type of the second blank?

⁷⁹ It must be *chain*, because *chain_item* consumes a pair consisting of an *int* and a *chain*.

What are our possibilities?

⁸⁰ The type of *Link*(*i*, *f*) is *chain*. But, since *f* is of type
 $int \rightarrow chain$,
f(*i*) is also of type *chain*.

Why do we use **and** to combine two definitions?¹

¹ You could also have written

```
local
  fun has_no_divisors(n,c)
    = if eq_int(c,1)
      then true
      else
        if divides_evenly(n,c)
        then false
        else has_no_divisors(n,c - 1)
  in
    fun is_prime(n)
      = has_no_divisors(n,n - 1)
end
...
```

⁹⁴ Because the first definition is more important than the second one and yet it refers to the second one.

¹ ...or

```
fun is_prime(n)
  = let
    fun has_no_divisors(n,c)
      = if eq_int(c,1)
        then true
        else
          if divides_evenly(n,c)
          then false
          else has_no_divisors(n,c - 1)
    in
      has_no_divisors(n,n - 1)
  end
```

Do we now know what the types of *is_prime* and *has_no_divisors* are?

⁹⁵ Now it is easy because we know that *has_no_divisors* produces a *bool*.

is_prime :
int → *bool*

•

has_no_divisors :
(*int* * *int*) → *bool*

•

Here is another long chain link.

⁹⁶ 37.

```
fun primes(n)
  = if is_prime(n + 1)
    then Link(n + 1,primes)
    else primes(n + 1)
```

primes :
int → *chain*

•

What is the value of
chain_item(12,*primes*(1))?

Do you like rabbits?

⁹⁷ Perhaps not to eat but to pet.

Here is how to make more rabbits.¹

```
fun fibs(n)(m)
  = Link(n + m, fibs(m))
```

What does *fibs* consume?

¹ In the *Liber abaci*, Leonardo of Pisa (1175–1250), also known as Fibonacci, describes the following problem. A pair of rabbits is placed in a pen to find out how many offspring will be produced by this pair in one year if each pair of rabbits gives birth to a new pair of rabbits each month starting with the second month of its life. The solution is known as the Fibonacci Sequence of numbers.

⁹⁸ It seems to consume two things, called *n* and *m*. Since we add them together, they must be *ints*.

What does *fibs* produce?

⁹⁹ That's easier. It produces a chain.

So why isn't this the type of *fibs*?

```
fibs:
(int * int) → chain
```



¹⁰⁰ Because there is no comma between *n* and *m*; instead there is *)*(.

What must be the type of *fibs*(*n*)?

¹⁰¹ We know that *Link* consumes an *int* and a function from *int* to *chain*. So, *fibs*(*n*) must be a function from *int* to *chain*.

So what is the type of *fibs*?

¹⁰² It really consumes just one *int*.

And then?

¹⁰³ It produces a function from *int* to *chain*.

So what is the type of *fibs*?

¹⁰⁴ Now it is obvious.

```
fibs :
int → (int → chain)
```



Yes, and we just found out about another notation for building functions that return functions.

¹⁰⁵ Yes, we did.

What is the value of
`Link(0, fibs(1))`?

¹⁰⁶ If you know this, take a short nap.

To determine its value, we only need to know the value of *fibs*(1).

¹⁰⁷ Yes, but what is it?

What type of thing is *fibs*(1)?

¹⁰⁸ It is a function of type *int* \rightarrow *chain*.

Here is such a function.

¹⁰⁹ It is like *fibs*, without (*n*).

```
fun fibs_1(m)  
  = Link(1 + m, fibs(m))
```

Where does it come from?

What showed up in place of *n*?

¹¹⁰ Every place where *n* appeared in the definition, except behind *fibs*, there is a 1 now.

We think of *fibs_1* as the value of *fibs*(1).

¹¹¹ That is simple enough.

What is the value of
`fibs(1)(1)`?

¹¹² The same as the value of
`fibs_1(1)`.

Do you see the underscores under the 1's?

¹¹³ Yes, and the 1 without an underscore has been consumed in the process.

What is the value of
fibs_1(1)?

¹¹⁴ `Link(2,fibs_1)`, a chain.

What is the value of
fibs_1(2)?

¹¹⁵ The same as the value of
`Link(3,fibs(2))`.

What is the value of
fibs(2)?

¹¹⁶ It is a function from *int* to *chain*.

Define *fibs_2*.

¹¹⁷ This is easy as pie.

```
fun fibs_2(m)  
  = Link(2 + m,fibs(m))
```

Don't forget the ice cream!

¹¹⁸ Okay.

The Seventh Moral

Some functions consume values of arrow type; some produce values of arrow type.

8. Bows and Arrows



Copyrighted image

Do you know *fish lists* and *int lists*?

¹ The datatype definition of *list* is an old family friend of ours.

```
datatype  $\alpha$  list =  
  Empty  
  | Cons of  $\alpha * \alpha$  list
```

Can you compare apples to oranges?

² No, how could we do that?

First, we put them together in a new datatype.

```
datatype orapl1 =  
  Orange  
  | Apple
```

³ Then comparing them is easy.

```
fun eq_orapl(Orange,Orange)  
  = true  
  | eq_orapl(Apple,Apple)  
  = true  
  | eq_orapl(one,another)  
  = false
```

```
eq_orapl :  
  (orapl * orapl)  $\rightarrow$  bool
```

¹ A better name for this is `orange_or_apple`.

Here is *subst_int*.

```
fun subst_int(n,a,Empty)  
  = Empty  
  | subst_int(n,a,Cons(e,t))  
  = if eq_int(a,e)  
    then Cons(n,subst_int(n,a,t))  
    else Cons(e,subst_int(n,a,t))
```

```
subst_int :  
  (int * int * int list)  $\rightarrow$  int list
```

⁴ It basically looks like *subst_int* and has a similar type.

```
fun subst_orapl(n,a,Empty)  
  = Empty  
  | subst_orapl(n,a,Cons(e,t))  
  = if eq_orapl(a,e)  
    then Cons(n,subst_orapl(n,a,t))  
    else Cons(e,subst_orapl(n,a,t))
```

```
subst_orapl :  
  (orapl * orapl * orapl list)  $\rightarrow$  orapl list
```

Define *subst_orapl*.

Would *subst_bool* be more difficult to define? ⁵ No, we would have to substitute *bool* for *int* everywhere in *subst_int*.

Would *subst_num* be more difficult to define? ⁶ No, we would have to substitute *num* for *int* everywhere in *subst_int*.

Would *subst_fish* be more difficult to define? ⁷ No, we would have to substitute *fish* for *int* everywhere in *subst_int*.

Are you tired of all this duplication yet? ⁸ Yes, is this going somewhere?

Okay, so let's not duplicate this work over and over again. ⁹ It is a function that consumes a value with four components, and that's what we can see immediately:

```
fun subst(rel1, n, a, Empty)
  = Empty
| subst(rel, n, a, Cons(e, t))
  = if rel(a, e)
    then Cons(n, subst(rel, n, a, t))
    else Cons(e, subst(rel, n, a, t))
```

(_____ * _____ * _____ * _____) → _____

What is the type of *subst*?

¹ A better name for *rel* is *relation*.

What do we know about the last component that *subst* consumes? ¹⁰ It must be a list, but since we don't know what kind of elements the list contains, we use α *list*:

(_____ * _____ * _____ * α *list*) → _____

How is the type of the result related to that of the fourth component? ¹¹ If *rel* always produced *false*, then the answer would have to be identical to the fourth component consumed. So, the type on the right of \rightarrow must be α *list*:

(_____ * _____ * _____ * α *list*) → α *list*.

What does α mean here?

¹² If *subst* consumes an *int list*, it produces an *int list*; if it consumes a *num list*, it produces a *num list*; and if it consumes a (*num list*) *list*, it produces a (*num list*) *list*; and if it consumes an *orapl list*, it produces an *orapl list*.

Does that imply anything else?

¹³ Yes, since *exp*₂ and *exp*₃ in
 if *exp*₁
 then *exp*₂
 else *exp*₃
are of the same type, this also means that *n* and *e* are of the same type. Since *e* is an element of the consumed list and is therefore of type α , so is *n*:
 (_____ * α * _____ * α *list*) \rightarrow α *list*.

Does that mean the type of *a* is α ?

¹⁴ Who knows? We don't know what *rel* consumes.

Does that mean we don't know what kind of value it is?

¹⁵ Yes, and so we could agree that its type is β :
 (_____ * α * β * α *list*) \rightarrow α *list*.

When is α different from β ?

¹⁶ On occasion, β will stand for the same type as α , and sometimes it will be a different type.

What is the type of *rel*?

¹⁷ It is a function that obviously produces *bool* and consumes a β and an α . And we don't know anything more about its type.

subst :
 (((β * α) \rightarrow *bool*) * α * β * α *list*)
 \rightarrow α *list*

•

Describe in your words what that type says about *subst*.

¹⁸ You knew that we would use *our* words:
“The type says that *subst* consumes a value with four components: a function, an arbitrary value of type α , another arbitrary value of type β , and a list. But, all elements in the list must have the type α , and the function must consume pairs of type $\beta * \alpha$.”

Anything else?

¹⁹ Of course, the result of *subst* is a list whose elements are of the same type as the first arbitrary value.

Suppose we want to substitute one *int* in a list of *ints* by some other *int*.

²⁰ Then, we need to give *subst* a function that consumes two *ints* as its first argument.

Do we know of such a function?

²¹ Yes, we do: *eq_int*.

So how do we use *subst* to substitute all occurrences of 15 in

```
Cons(15,
  Cons(6,
    Cons(15,
      Cons(17,
        Cons(15,
          Cons(8,
            Empty))))))
```

by 11?

²² We use *eq_int* as *rel* and otherwise act as if we were using *subst_int*:

```
subst(eq_int, 11, 15,
  Cons(15,
    Cons(6,
      Cons(15,
        Cons(17,
          Cons(15,
            Cons(8,
              Empty)))))))).
```

And that produces?

²³ A list with three 11's.

```
Cons(11,
  Cons(6,
    Cons(11,
      Cons(17,
        Cons(11,
          Cons(8,
            Empty))))))).
```

What is the value of
less_than(15,17)?

²⁴ true.

Is *less_than* a function that consumes a
two-component value with both components
being *ints*?

²⁵ Yes, that's right.

Can we use it with *subst*?

²⁶ Yes, we can substitute all *ints* in an *int list*
that are greater than or equal to some other
int.

So how would we substitute all numbers not
less than 15 in

```
Cons(15,  
  Cons(6,  
    Cons(15,  
      Cons(17,  
        Cons(15,  
          Cons(8,  
            Empty))))))
```

by 11?

²⁷ We use *less_than* as *rel* and otherwise act as
if we were using *subst_int*:

```
subst(less_than,11,15,  
  Cons(15,  
    Cons(6,  
      Cons(15,  
        Cons(17,  
          Cons(15,  
            Cons(8,  
              Empty)))))))).
```

And what does that produce?

²⁸ A list with an 11:

```
Cons(15,  
  Cons(6,  
    Cons(15,  
      Cons(11,  
        Cons(15,  
          Cons(8,  
            Empty))))))).
```

What is the value of
in_range((11,16),15)?

²⁹ true.

What does *in_range* do?

³⁰ It determines whether or not some number is
in some range of numbers.

And what is *pred*?

⁴⁴ It is a function that consumes one value, an element of the list, and produces a *bool*.

subst_pred :
 $((\alpha \rightarrow \text{bool}) * \alpha * \alpha \text{ list}) \rightarrow \alpha \text{ list}$

•

Describe in your words what that type says about *subst_pred*.

⁴⁵ Here are our words again:
“The type says that *subst_pred* consumes a value with three components: a function, an arbitrary value of type α , and a list. But, all elements in the list must have type α , and the function must consume values of that type.”

Anything else?

⁴⁶ Same as before. The result of *subst_pred* is a list whose elements are of the same type as the arbitrary value.

So how do we use *subst_pred* to substitute all occurrences of 15 in

```
Cons(15,  
  Cons(6,  
    Cons(15,  
      Cons(17,  
        Cons(15,  
          Cons(8,  
            Empty))))))
```

by 11?

⁴⁷ We need a function that compares the value it consumes to 15.

Define this function.

⁴⁸ Easy.

fun *is_15*(*n*)
 = *eq_int*(*n*,15)

is_15 :
int \rightarrow *bool*

•

Does *in_range* consume an *int*?

⁶⁰ That's what its type says.

Can we use it with *subst_pred*?

⁶¹ Well, we could as long as the third component consumed by *subst_pred* is an *int list*.

So how would we substitute all numbers between 11 and 16 in

```
Cons(15,
  Cons(6,
    Cons(15,
      Cons(17,
        Cons(15,
          Cons(8,
            Empty))))))
```

by 22?

⁶² We use *in_range_11_16* as *pred*:
subst_pred(in_range_11_16,22,
 Cons(15,
 Cons(6,
 Cons(15,
 Cons(17,
 Cons(15,
 Cons(8,
 Empty))))))).

And what does that produce?

⁶³ A list with three 22's:
Cons(22,
 Cons(6
 Cons(22,
 Cons(17,
 Cons(22,
 Cons(8,
 Empty))))))).

We recommend dinner now. How about some Indian lamb?

⁶⁴ Don't forget the curry.

Did you have your fill of curry? Then take a look at this variant of *in_range_11_16*.

```
fun in_range_c(small,large)(x)
  = if less_than(small,x)
    then less_than(x,large)
    else false
```

What is different about it besides its name?

⁶⁵ It is like *in_range_11_16*, but it doesn't contain 11 and 16. Instead, it first consumes a pair of *ints* and then another *int*.¹

¹ Such functions are said to be *curried*. A better name for this function would be *in_range_Curry* after Haskell B. Curry (1900–1982) and Moses Schönfinkel (1889–1942).

So what is the type of *in_range_c*?

⁶⁶ We need to substitute just one *** with an *→* in the type of *in_range*.

```
in_range_c :  
  (int * int) → int → bool
```

•

What is the purpose of the underlined parentheses?

⁶⁷ They surround the type of what *in_range_c* produces.

Does that mean that *in_range_c* is a function that consumes one pair of *ints*?

⁶⁸ Yes, and it produces a function.

What does the function that it produces consume?

⁶⁹ That function consumes an *int*, just like *in_range_11_16*.

What is the value of
in_range_c(11,16)?

⁷⁰ It is a function, and that function is just like *in_range_11_16*.

Can you define a function that is like
in_range_c(11,16)?

⁷¹ We copy *in_range_c* and substitute 11 for *small* and 16 for *large*.

```
fun in_range_c_11_16(x)  
  = if less_than(11,x)  
    then less_than(x,16)  
    else false
```

So what is the difference between
in_range_11_16
and
in_range_c_11_16?

⁷² None.

What is the value of

```
combine(  
  Cons(1,  
    Cons(2,  
      Cons(3,  
        Empty))),  
  Cons(5,  
    Cons(4,  
      Cons(7,  
        Cons(9,  
          Empty))))))?
```

⁹⁰ That's no problem:

```
Cons(1,  
  Cons(2,  
    Cons(3,  
      Cons(5,  
        Cons(4,  
          Cons(7,  
            Cons(9,  
              Empty)))))))).
```

What is the value of

```
combine(  
  Cons(1,  
    Cons(2,  
      Cons(3,  
        Empty))),  
  Cons(12,  
    Cons(11,  
      Cons(5,  
        Cons(7,  
          Empty))))))?
```

⁹¹ It starts with the same numbers:

```
Cons(1,  
  Cons(2,  
    Cons(3,  
      Cons(12,  
        Cons(11,  
          Cons(5,  
            Cons(7,  
              Empty)))))))).
```

Define *combine_c*.

⁹² That must be the function that consumes one list and produces a function that consumes a list and then produces the combined list.

Yes.

⁹³ That's easy then.

```
fun combine_c(Empty)(l2)  
  = l2  
  | combine_c(Cons(a,l1))(l2)  
    = Cons(a,combine_c(l1)(l2))
```

```
combine_c :  
  α list → (α list → α list)
```

•

What is the value of
combine_s(
 Cons(1,
 Cons(2,
 Cons(3,
 Empty))))?)

¹¹⁶ It is equivalent to the value of
make_cons(1
 make_cons(2,
 make_cons(3,
 base))).

What is the value of
make_cons(3,
 base)?

¹¹⁷ It is this function.

```
fun prefix_3(l2)
  = Cons(3,base(l2))
```

```
prefix_3 :
  int list → int list
```

•

Then what is the value of
make_cons(2,
 prefix_3)?

¹¹⁸ No big deal.

```
fun prefix_23(l2)
  = Cons(2,prefix_3(l2))
```

```
prefix_23 :
  int list → int list
```

•

So what is the value of
make_cons(1,
 prefix_23)?

¹¹⁹ A function that consumes a list and prefixes
that list with 1, 2, and 3.

```
fun prefix_123(l2)
  = Cons(1,prefix_23(l2))
```

```
prefix_123 :
  int list → int list
```

•

Is *prefix_123* equal to *prefixer_123*?

¹²⁰ Extensionally, yes. Both prefix a list with 1, 2, and 3. Intensionally, no. The latter just Conses the numbers onto a list, but the former has to shuffle the list around with *make_cons*.

Can we define a function like *combine_s* that produces *prefixer_123* when used with

```
Cons(1,  
    Cons(2,  
        Cons(3,  
            Empty))))?
```

¹²¹ We'd rather have dessert. How about you?

What is the difference between functions of type

$\text{type}_1 \rightarrow \text{type}_2 \rightarrow \text{type}_3$

and those of type

$(\text{type}_1 * \text{type}_2) \rightarrow \text{type}_3$?

¹²² Easy, they have different types.

Seriously.

¹²³ The first kind of function consumes two values in two stages and may determine some aspect of the value it produces before it consumes the second value. The second kind of function consumes two values as a pair.

Aren't functions a lot of **fun**?

¹²⁴ They sure are.

Rest up before continuing, unless you are exceptionally hungry.

¹²⁵ See you tomorrow.

The Eighth Moral

Replace stars by arrows to reduce the number of values consumed and to increase the generality of the function defined.

Did you ever play “Steal the Bacon?”

¹ No, what about it?

We just invented “Find the Bacon.”

² How does it work?

We need to practice first.

³ What are we waiting for?

Lists.

⁴ Lists are easy, they have been done before.

```
datatype  $\alpha$  list =  
  Empty  
  | Cons of  $\alpha * \alpha$  list
```

And we also use this datatype.

⁵ There is some Bacon.

```
datatype box1 =  
  Bacon  
  | lx1 of int
```

¹ Better names for these are `bacon_or_index` and `Index`, respectively.

What is the value of

⁶ 3, right?

```
where_is(  
  Cons(lx(5),  
    Cons(lx(13),  
      Cons(Bacon,  
        Cons(lx(8),  
          Empty))))))?
```

What is the value of

⁷ 1, because Bacon is the first thing in the list.

```
where_is(  
  Cons(Bacon,  
    Cons(lx(8),  
      Empty))))?
```

Oh No!

What should be the value of

```
where_is(  
  Cons(lx(5),  
    Cons(lx(13),  
      Cons(lx(8),  
        Empty)))))?
```

⁸ 0, because there is no Bacon in the list.

Here is the function *is_bacon*.

```
fun is_bacon(Bacon)  
  = true  
  | is_bacon(lx(n))  
  = false
```

```
is_bacon :  
  box → bool
```

Use it to define *where_is*.

⁹ This shouldn't be a problem.

```
fun where_is(Empty)  
  = 0  
  | where_is(Cons(a_box,rest))  
  = if is_bacon(a_box)  
    then 1  
    else 1 + where_is(rest)
```

```
where_is :  
  box list → int
```

Use your definition to determine the value of

```
where_is(  
  Cons(lx(5),  
    Cons(lx(13),  
      Cons(lx(8),  
        Empty))))).
```

¹⁰ Oh no. It's 3.

What were we expecting?

¹¹ 0, of course.

How did that happen?

¹² When *where_is* produced 0, three additions were waiting for the result:

```
1 +  
1 +  
1 + ...
```

We should forget these additions when we return 0, shouldn't we?

¹³ That would be great.

There is a way to do precisely that. Take a look at these definitions.

```
exception No_bacon of int
```

```
fun where_is(Empty)
  = raise No_bacon(0)
| where_is(Cons(a_box, rest))
  = if is_bacon(a_box)
    then 1
    else 1 + where_is(rest)
```

¹⁴ They contain two new special words:
exception and **raise**.

Use your own words to describe what **exception** means.

¹⁵ You knew that we wouldn't let you down. Here are our words:
"The **exception** definition creates a constructor just like a **datatype** definition but for exceptional values. The expression `No_bacon(10)` creates such an exceptional value."

What does **raise** mean?

¹⁶ Can we just watch it in action?

Yes, let's slowly determine the value of

```
where_is(
  Cons(lx(5),
    Cons(lx(13),
      Cons(lx(8),
        Empty))))).
```

¹⁷ Yes, let's do that.

Since the list is constructed with `Cons`, this is equal to

```
1 + where_is(
  Cons(lx(13),
    Cons(lx(8),
      Empty))))).
```

¹⁸ Which in turn is equal to

```
1 +
  1 + where_is(
    Cons(lx(8),
      Empty)).
```

Correct. And that is equal to

```
1 +  
1 +  
1 + where_is(  
    Empty).
```

¹⁹ Now **raise** happens:

```
1 +  
1 +  
1 + raise No_bacon(0).
```

But what does that mean? Isn't it nonsense?

No,
because we can think of **raise** ... as
having the *necessary* type, whatever it
may be.

²⁰ We have never seen anything like that before.
Still, the answer does not explain what the
expression means.

The meaning of **raise** ... is equally simple.
It does not have a value.

²¹ It has no value? No wonder it has whatever
type it needs to have. But that is strange.

So next we have
raise No_bacon(0).

²² They've disappeared.

Where did the additions go?

Good. Now we know that
where_is(
 Cons(lx(5),
 Cons(lx(13),
 Cons(lx(8),
 Empty))))

²³ Does that mean we didn't make any
progress?

has no value but is equal to
raise No_bacon(0).

We made some progress. The additions are
gone.

²⁴ Yes, but we wanted an *int*; we wanted 0.

Don't worry, we will get there. Did you
notice that we said *where_is* does not
produce a value when it consumes a list
without Bacon?

²⁵ Yes, we said that.

But didn't we say that *where_is* produces an *int*?

²⁶ Yes, but how can we say that it doesn't produce an *int* for everything that it consumes?

We can't. We just know that when we say *where_is* is of type

$(\text{box list}) \rightarrow \text{int}$,

we include the possibility that the function **raises** an exception.

²⁷ Aha, that clarifies it.

Is that all that bad?

²⁸ It depends. If we only determine the value of

```
where_is(  
  Cons(lx(5),  
    Cons(lx(13),  
      Cons(lx(8),  
        Empty))))),
```

we are just fine. If we get a number, we know that the list contains **Bacon** and where it is. If it **raises** an exception, we know there is no **Bacon**.

How do we know when the function **raises** an exception?

²⁹ Good question.

We need yet another ingredient called **handle**.

³⁰ And how do we use this new ingredient?

Like this:

```
(where_is(  
  Cons(lx(5),  
    Cons(lx(13),  
      Cons(lx(8),  
        Empty))))  
handle  
  No_bacon(an_int)  
   $\Rightarrow^1$  an_int).
```

³¹ It seems like we are looking at a new form of expression:

$(\text{exp}_1 \text{ **handle** pattern} \Rightarrow \text{exp}_2)$.

But what does it mean?

¹ This is a two-character symbol: \Rightarrow .

What do you think it means?

³² We know that the **handle** expression consumes exceptional values. So, when

```
where_is(  
  Cons(lx(5),  
    Cons(lx(13),  
      Cons(lx(8),  
        Empty))))
```

is the same as

```
raise No_bacon(0),
```

it matches the **handler** pattern and produces whatever is to the right of \Rightarrow .

And how does `No_bacon(0)` match `No_bacon(an_int)`?

³³ That's barely worth a question. It's certainly not worthy of an answer.

Let *an_int* stand for 0. Then what is the value in our example?

³⁴ Exactly what we want: 0, which is what is to the right of \Rightarrow with *an_int* replaced by 0.

What is the value of

```
(where_is(  
  Cons(lx(5),  
    Cons(Bacon,  
      Cons(lx(8),  
        Empty))))  
handle  
  No_bacon(an_int)  
   $\Rightarrow$  an_int)?
```

³⁵ It is 2,
because **Bacon** is in the second position and no exception is **raised**.

What kind of value does

```
(where_is(  
  ...)  
handle  
  No_bacon(an_int)  
   $\Rightarrow$  an_int)
```

³⁶ An *int*.

produce if the value consumed contains **Bacon**?

Put in your own words what it means to say
some function f is of type

_____ $\rightarrow out$.

⁶¹

We say:

“If f produces a value, that value is of type
 out . But, the use of f may be meaningless
or it may **raise** an exception.”

Does every function type have this extended
meaning?

⁶²

Absolutely.

Time to define *find*, isn't it?

⁶³

Don't we need a function like *chain_item* for
lists?

Good point. Define it.

⁶⁴

Here is a part of it.

```
fun list_item(n, Empty)
  = _____
  | list_item(n, Cons(abox, rest))
  = if eq_int(n, 1)
    then abox
    else list_item(n - 1, rest)
```

Why is the first answer a blank?

⁶⁵

Because it is not clear what *list_item*
produces when the list is empty.

Let's raise an exception. Here is its
definition.

```
exception Out_of_range
```

⁶⁶

Well, then it is easy to fill in the blank.

```
fun list_item(n, Empty)
  = raise Out_of_range
  | list_item(n, Cons(abox, rest))
  = if eq_int(n, 1)
    then abox
    else list_item(n - 1, rest)
```

```
list_item :
  (int * box list)  $\rightarrow$  box
```

Does this definition differ from anything we have seen before?

⁶⁷ Very.

```
fun
  find(n,boxes)
    = check(n,boxes,list_item(n,boxes))
and
  check(n,boxes,Bacon)
    = n
  | check(n,boxes,lx(i))
    = find(i,boxes)
```

```
find :
  (int * (box list)) → int
```

```
check :
  (int * (box list) * box) → int
```

That's correct. Does the definition of *box* refer to itself?

⁶⁸ No.

Does the definition of *find* refer to itself?

⁶⁹ Yes, through *check*.

Isn't that unusual?

⁷⁰ We have not seen that combination before.

Does that mean the definition of *find* matches neither the outline of the datatype *box* nor that of the datatype *box list*?

⁷¹ That's right, it doesn't.

Then what is the reference to *find* used for?

⁷² It is used to restart the search for Bacon with a new index.

Isn't this unusual?

⁷³ Very.

Oh No!

And that kind of reference is precisely why a ⁷⁴ That settles it.
use of *find* may be meaningless.

What is the value of

find(1,*t*)

where *t* is

```
Cons(Ix(5),  
    Cons(Ix(4),  
        Cons(Bacon,  
            Cons(Ix(2),  
                Cons(Ix(7),  
                    Empty))))))?
```

⁷⁵ Is *t* going to change?¹

¹ We can write
val *t* =
 Cons(Ix(5),
 Cons(Ix(4),
 Cons(Bacon,
 Cons(Ix(2),
 Cons(Ix(7),
 Empty))))))

in order to associate the name *t* with this value.

No, it will stay the same for the rest of the chapter. So what is the value?

⁷⁶ The expression is the same as the value of
find(5,*t*).

And then?

⁷⁷ Then it is the same as *find*(7,*t*).

And now?

⁷⁸ An exception is raised.

And what does that mean?

⁷⁹ Every time *find* raises an exception, the bacon can't be found.

Let's try something new. We will restart the search at *n* div 2.

⁸⁰ What does that mean?

What is the value of 8 div 2?

⁸¹ Obvious: 4.

What is the value of 7 div 2?

⁸² Not so obvious: 3.

How can we restart the search when the number is out of range?

⁸³ We can use a **handler**.

Good. Fill in the blank.

```
fun find(n,boxes)
  = (check(n,boxes,list_item(n,boxes))
     handle
     Out_of_range
     ⇒ _____ )
and
  check(n,boxes,Bacon)
  = n
  | check(n,boxes,lx(i))
  = find(i,boxes)
```

⁸⁴ Okay.

```
fun find(n,boxes)
  = (check(n,boxes,list_item(n,boxes))
     handle
     Out_of_range
     ⇒ find(n div 2,boxes))
and
  check(n,boxes,Bacon)
  = n
  | check(n,boxes,lx(i))
  = find(i,boxes)
```

```
find :
  (int * (box list)) → int
```

```
check :
  (int * (box list) * box) → int
```

Now the plot really thickens.

⁸⁵ Like pea soup?

Now what is the value of
find(1,*t*)?

⁸⁶ It is the same as the value of
find(5,*t*)
handle
Out_of_range
⇒ *find*(1 div 2,*t*)).

And then?

⁸⁷ Then it is the same as
find(7,*t*)
handle
Out_of_range
⇒ *find*(5 div 2,*t*)
handle
Out_of_range
⇒ *find*(1 div 2,*t*)).

Oh No!

145

And now?

⁸⁸ The next stop is
(((*check*(7,*t*,*list_item*(7,*t*))
 handle
 Out_of_range
 ⇒ *find*(7 div 2,*t*))
 handle
 Out_of_range
 ⇒ *find*(5 div 2,*t*))
 handle
 Out_of_range
 ⇒ *find*(1 div 2,*t*)).

And here *list_item*(7,*t*) raises an exception.

What does that mean?

⁸⁹ It means *list_item*(7,*t*) doesn't have a value but is equal to **raise** Out_of_range, so that we get

```
(((check(7,t,raise Out_of_range)
  handle
  Out_of_range
  ⇒ find(7 div 2,t)
  handle
  Out_of_range
  ⇒ find(5 div 2,t)
  handle
  Out_of_range
  ⇒ find(1 div 2,t)).
```

Does *check*(7,*t*,...) disappear, too?

⁹⁰ Yes, **raise** does that.

How is the exception **handled** then?

⁹¹ By matching with Out_of_range.

Yes, and then we evaluate *find*(7 div 2,*t*).
What is the next expression?

⁹² Easy:
((*find*(3,*t*)
 handle
 Out_of_range
 ⇒ *find*(5 div 2,*t*)
 handle
 Out_of_range
 ⇒ *find*(1 div 2,*t*)).

Next?

⁹³ We have found the **Bacon**, which means the result is 3.

Where have the **handlers** gone?

⁹⁴ Since $find(3,t)$ has a value, the **handlers** disappear.

Where did we stop while we were searching for the bacon?

⁹⁵ At 1, 5, 7, and 3.

Could we define a function that produces that sequence for us?

⁹⁶ Yes, as an *int list*.

Hang on!

⁹⁷ Is it going to get more complicated still?

Yes! Look at this definition of *path*.

⁹⁸ No, that much is obvious.

```
fun path(n,boxes)
  = Cons(n,
    (check(n,boxes,list_item(n,boxes))
     handle
      Out_of_range
      ⇒ path(n div 2,boxes)))
and
  check(n,boxes,Bacon)
  = Empty
  | check(n,boxes,lx(i))
  = path(i,boxes)
```

```
fun path(n,boxes)
  = Cons(n,
    (check(boxes,list_item(n,boxes))
     handle
      Out_of_range
      ⇒ path(n div 2,boxes)))
and
  check(boxes,Bacon)
  = Empty
  | check(boxes,lx(i))
  = path(i,boxes)
```

```
path :
  (int * (box list)) → (int list) •
```

```
path :
  (int * (box list)) → (int list) •
```

```
check :
  (int * (box list) * box) → (int list) •
```

```
check :
  ((box list) * box) → (int list) •
```

Do we still need to have *n* around in *check*?

Oh No!

Describe in your own words how this function⁹⁹ What?
produces the list of intermediate stops.

Neither can we. So let's just determine the
value of

path(1,*t*).

¹⁰⁰ Well, *list_item* produces *lx*(5), which means
that it is equal to

Cons(1,
 (*path*(5,*t*)
 handle
 Out_of_range
 \Rightarrow *path*(1 div 2,*t*))).

And then?

¹⁰¹ Then it is the same as

Cons(1,
 (*Cons*(5,
 (*path*(7,*t*)
 handle
 Out_of_range
 \Rightarrow *path*(5 div 2,*t*)))
 handle
 Out_of_range
 \Rightarrow *path*(1 div 2,*t*))).

And the next stop is

Cons(1,
 (*Cons*(5,
 (*Cons*(7,
 (*check*(*t*,*list_item*(7,*t*))
 handle
 Out_of_range
 \Rightarrow *path*(7 div 2,*t*)))
 handle
 Out_of_range
 \Rightarrow *path*(5 div 2,*t*)))
 handle
 Out_of_range
 \Rightarrow *path*(1 div 2,*t*))).

Here *list_item*(7,*t*) raises an exception.

¹⁰² Right and we get

Cons(1,
 (*Cons*(5,
 (*Cons*(7,
 (*check*(*t*,**raise** *Out_of_range*)
 handle
 Out_of_range
 \Rightarrow *path*(7 div 2,*t*)))
 handle
 Out_of_range
 \Rightarrow *path*(5 div 2,*t*)))
 handle
 Out_of_range
 \Rightarrow *path*(1 div 2,*t*))).

What is the value of *plus*(0,1)?

¹ 1. But isn't that obvious?

Yes, it's obvious, so let's move on. What is the value of *plus*(1,1)?

² 2, which is also one more than *plus*(0,1).

Correct. Here is the final question. What is the value of *plus*(2,1)?

³ 3, which is one more than *plus*(1,1).

Here is a definition of *plus* based on the previous questions.

```
fun plus(n,m)
  = if is_zero(n)
    then m
    else succ(plus(pred(n),m))
```

```
plus :
  (int * int) → int
```

⁴ They are easy functions.

```
fun is_zero(n)
  = eq_int(n,0)
```

```
is_zero :
  int → bool
```

```
exception Too_small
```

```
fun pred(n)
  = if eq_int(n,0)
    then raise Too_small
    else n - 1
```

```
pred :
  int → int
```

```
fun succ(n)
  = n + 1
```

```
succ :
  int → int
```

It relies on three help functions: *is_zero*, *pred*, and *succ*.¹ Define these help functions.

¹ Better names for these functions are *predecessor* and *successor*, respectively.

Why does *pred* raise an exception when it consumes 0?

⁵ We only work with non-negative *ints*, so 0 does not have a predecessor.

Define the function *plus* in the same style, but use *nums* in place of *ints*.

⁶ With those, it is a piece of cake.

```
datatype num =  
  Zero  
| One_more_than of num
```

```
fun plus(n,m)  
  = if is_zero(n)  
    then m  
    else succ(plus(pred(n),m))
```

Here are the help functions that we need.

```
fun is_zero(Zero)  
  = true  
| is_zero(not_zero)  
  = false
```

```
plus :  
  (num * num) → num
```

```
is_zero :  
  num → bool
```

```
exception Too_small
```

```
fun pred(Zero)  
  = raise Too_small  
| pred(One_more_than(n))  
  = n
```

```
pred :  
  num → num
```

```
fun succ(n)  
  = One_more_than(n)
```

```
succ :  
  num → num
```

Isn't it curious that the two definitions of *plus* are identical?

⁷ Yes, and that's good.

Why?

⁸ Because the functions are closely related. They produce similar values when they consume similar pairs of values.

What is the value of *plus*(2,3)?

⁹ This is nonsense. The last definition of *plus* consumes a pair of *nums* and produces one. It cannot be used with *ints*.

What is the value of *plus*(
 One_more_than(
 One_more_than(
 Zero)),
 One_more_than(
 One_more_than(
 One_more_than(
 Zero))))?

¹⁰ Now we are making sense. It is
 One_more_than(
 One_more_than(
 One_more_than(
 One_more_than(
 One_more_than(
 Zero))))).

Isn't it unfortunate that we can't use the two versions of *plus* at the same time?

¹¹ It truly is. But because the two definitions are identical, we must use building blocks with the same names, even though they consume and produce values of different types.

Any ideas about what to do?

¹² There seems to be no other way to do this. For each definition of *plus* we need to have around the two sets of building blocks. Each set requires definitions for the same set of names. Because it is impossible to use a name for two different definitions, we cannot have two definitions of *plus* available at the same time.

There is a way and we are about to discover it.

¹³ Oh great.

What are the basic building blocks needed to make *plus*?

¹⁴ There are five:
the type,
the exception *Too_small*,
the function *succ*,
the function *pred*,
and
the function *is_zero*.

If we call the type *number*, what is the type of the building block *succ*?

¹⁵ The type of *succ* is
number \rightarrow *number*.

And how about *pred*?

¹⁶ It has the same type:
number \rightarrow *number*.

And *is_zero*?

¹⁷ It produces a *bool*:
number \rightarrow *bool*.

Good, and here is a way to write down these minimal requirements for our building blocks.

```
signature N1 =  
sig  
  type number  
  exception Too_small  
  val succ : number  $\rightarrow$  number  
  val pred : number  $\rightarrow$  number  
  val is_zero : number  $\rightarrow$  bool  
end
```

¹⁸ This is clear enough. This notation specifies five things between **sig** and **end**, but what do **signature**, **type**, and **val** mean?

¹ A better name for this signature would be `NUMBERS_BY_PEANO`.

What does `()` mean?

²⁹ Here are our words:
“It means that we are *defining* a functor that does not depend on anything else.”

Good. We will see things other than `()`.

³⁰ Okay, and then the meaning of “depend” should become clearer.

So what is the notation `... ▷ N` about?

³¹ It states that the result of using the functor is a structure with signature *N*.

Do both of these functors produce structures that have the signature *N*?

³² Each **struct** ... **end** contains several definitions, but at least one for *number*, *Too_small*, *succ*, *pred*, and *is_zero*. And, in terms of *number*, the three **values** have the right type.

Now let’s use a functor to build a structure.

```
structure IntStruct =  
  NumberAsInt()
```

³³ It is *N* obviously, because the definition of *NumberAsInt* states that the functor produces structures with signature *N*.

What is the signature of *IntStruct*?

And what does `()` behind *NumberAsInt* mean?

³⁴ Here are our words:
“It means that we are *using* a functor that does not depend on anything else.”

Define the structure *NumStruct*.

³⁵ That’s obvious now.

```
structure NumStruct =  
  NumberAsNum()
```

Why are we doing all of this?

³⁶ Is it because we want to use both versions of *plus* at the same time and, if possible, create them from the same text?

Do we now have both sets of building blocks around at the same time?

³⁷ Basically. Those for *nums* are collected in *NumStruct* and those for *ints* in *IntStruct*.

Is this progress?

³⁸ Yes, if we can now somehow create the two versions of *plus* from the two structures.

Exactly. What is the type of *plus*?

³⁹ If *number* is the type, then *plus* has the type $(number * number) \rightarrow number$.

Define a signature that says that.

⁴⁰ Here is one.

```
signature P1 =  
sig  
  type number  
  val plus :  
    (number * number) → number  
end
```

¹ A better name for this signature would be `PLUS_OVER_NUMBER`.

Here is the functor.

```
functor PON1(structure a_N : N)  
▷  
P  
=  
struct  
  type number = a_N.number  
  fun plus(n,m)  
    = if a_N.is_zero(n)  
      then m  
      else a_N.succ(  
        plus(a_N.pred(n),m))  
end
```

⁴¹ The names of the other functors are always followed by (). This one, however, contains something else:
(structure a_N : N).
What does it mean?

How does it differ from the functors we have seen so far?

¹ A better name for this functor would be `PlusOverNumber`.

The notation

(structure $a_N : N$)

says that the structure produced by PON depends on a structure a_N that has signature N .

⁴² And that's how we know that a_N contains a type, an exception, and three values: *is_zero*, *succ*, and *pred*.

What does $a_N.is_zero$ mean?

⁴³ It means that we are using the **value** named *is_zero* from the structure named a_N .

Correct. And how about

$a_N.number$,

$a_N.succ$,

and

$a_N.pred$?

⁴⁴ They refer to
 a_N 's **type** *number*,
 a_N 's **value** *succ*,
and
 a_N 's **value** *pred*,
respectively.

And how do we know that a_N contains all these things?

⁴⁵ Because it has signature N .

Let's build a structure from PON .

⁴⁶ We don't know how to satisfy PON 's dependency.

We need a new notation.

⁴⁷ Yet more notation?

structure *IntArith* =
 $PON(\text{structure } a_N = \text{IntStruct})$

Yes. Explain in your words what it means.

⁴⁸ Our words:
"Consider the functor's dependency:
(structure $a_N : N$).
It specifies that the structure created by PON depends on a yet to be determined structure a_N with signature N . Here we say that a_N stands for *IntStruct*."

Does *IntStruct* have the signature *N*?

⁴⁹ The structure was created with *NumberAsInt*, which always produces structures that have signature *N*.

And how do we know that?

⁵⁰ The definition of *NumberAsInt* contains *N* below \triangleright , and that's what says the resulting structure has signature *N*.

Time to create *plus* over *nums*.

⁵¹ Easy.

```
structure NumArith =  
  PON(structure a_N = NumStruct)
```

What is the value of
IntArith.plus(1,2)?

⁵² This should be 3.

Wrong.

⁵³ What! Nonsense!

Good guess! It is nonsense. What do we know about *IntArith*?

⁵⁴ We know that it is a structure that has signature *P*.

What do we know about structures that have signature *P*?

⁵⁵ A structure with signature *P* has two components: a type named *number* and a value named *plus*. The value *plus* consumes a pair of *numbers* and produces one.

And what else do we know about *number* in *P*?

⁵⁶ Nothing, because the signature *P* does not reveal anything else about the structures that *PON* produces.

Absolutely. And that's why it is nonsense to ask for the value of
IntArith.plus(1,2).

⁵⁷ Okay, that's clear. The function *IntArith.plus* consumes values of type *IntArith.number*, about which *P* doesn't reveal anything, but 1 and 2 are *ints*.

Can we determine the value of
NumArith.plus(
 One_more_than(Zero),
 One_more_than(
 One_more_than(Zero)))?

⁵⁸ No. The function *NumArith.plus* consumes values of type *NumArith.number*, but
 One_more_than(Zero)
and
 One_more_than(
 One_more_than(Zero))
are *nums*.

Do we have the means to produce *numbers* of the correct type for either *IntArith.plus* or *NumArith.plus*?

⁵⁹ No, the two structures contain only one function, *plus*, and it assumes that we have *numbers* ready for consumption.

How about the structures *IntStruct* and *NumStruct*?

⁶⁰ They, too, provide only functions that consume existing *numbers*.

So what do we do?

⁶¹ Yes, what?

Here is one way out. Let's use a larger signature.

```
signature N_C_R =  
sig  
  type number  
  exception Too_small  
  val conceal : int → number  
  val succ : number → number  
  val pred : number → number  
  val is_zero : number → bool  
  val reveal : number → int  
end
```

⁶² The signature *N_C_R*¹ requires that its corresponding structures contain definitions for two additional functions: *conceal* and *reveal*. What can they be about?

¹ A better name for this signature would be *NUMBERS_WITH_CONCEAL_REVEAL*.

The function *conceal* consumes an *int* and produces a similar *number*.

⁶³ Does *reveal* do the opposite?

Yes, and opposite means that for any *int* $x \geq 0$,

$$\text{reveal}(\text{conceal}(x)) = x.$$

⁶⁴ Oh, *conceal* is like $(\cdot)^2$ (square) and *reveal* like $\sqrt{\cdot}$ (square root) because for any *int* $x \geq 0$,

$$\sqrt{x^2} = x.$$

Good. Here is the extended version of *NumberAsInt*.

```
functor NumberAsInt()
▷
N_C_R
=
struct
  type number = int
  exception Too_small
  fun conceal(n)
    = n
  fun succ(n)
    = n + 1
  fun pred(n)
    = if eq_int(n,0)
      then raise Too_small
      else n - 1
  fun is_zero(n)
    = eq_int(n,0)
  fun reveal(n)
    = n
end
```

Define the extended version of *NumberAsNum*.

⁶⁵ That requires a bit more thought.

```
functor NumberAsNum()
▷
N_C_R
=
struct
  datatype num =
    Zero
    | One_more_than of num
  type number = num
  exception Too_small
  fun conceal(n)
    = if eq_int(n,0)
      then Zero
      else One_more_than(
        conceal(n - 1))
  fun succ(n)
    = One_more_than(n)
  fun pred(Zero)
    = raise Too_small
    | pred(One_more_than(n))
    = n
  fun is_zero(Zero)
    = true
    | is_zero(a_num)
    = false
  fun reveal(n)
    = if is_zero(n)
      then 0
      else 1 + reveal(pred(n))
end
```

Let's rebuild the structures *IntStruct* and *IntArith*.

```
structure IntStruct =  
  NumberAsInt()
```

```
structure IntArith =  
  PON(structure a_N = IntStruct)
```

⁶⁶ Okay, here are the new versions of *NumStruct* and *NumArith*.

```
structure NumStruct =  
  NumberAsNum()
```

```
structure NumArith =  
  PON(structure a_N = NumStruct)
```

What kind of structures are *IntStruct* and *NumStruct*?

⁶⁷ Both have signature *N_C_R*.

What kind of structure does *PON* depend on?

⁶⁸ It depends on a structure with signature *N*. Isn't this a conflict?

Does a structure with signature *N_C_R* provide all the things that a structure with signature *N* provides?

⁶⁹ It does, and *N_C_R* even lists those pieces that are also in *N* in the same order as *N*.

Absolutely. And that's why it is okay to supply *IntStruct* and *NumStruct* to *PON*.

⁷⁰ Okay.

What is the value of
 NumStruct.reveal(
 NumStruct.succ(
 NumStruct.conceal(0)))?

⁷¹ 1,
because *NumStruct.conceal* consumes an *int* and produces a *number* for the consumption of *NumStruct.succ*. And *NumStruct.reveal* consumes a *number* and produces an *int*.

What is the value of
 NumStruct.reveal(
 NumArith.plus(
 NumStruct.conceal(1),
 NumStruct.conceal(2)))?

⁷² This should be 3, now.

Are there other forms of signature nonsense? ⁸¹ Here is one:

NumStruct.Zero.

The signature doesn't say anything about a constructor *Zero*, so we can't know anything about it either.

Correct.

⁸² What shall we do?

We need to say that *PON* produces structures whose type *number* is the same as the type *number* in *a_N*, the functor's dependency.

⁸³ And how do we do that?

We connect the signature of the structure produced by *PON* to the structure on which it depends.

⁸⁴ Is
P where type number = a_N.number
a signature?

```
functor PON(structure a_N : N)
▷
P where type number = a_N.number
=
struct
  type number = a_N.number
  fun plus(n,m)
    = if a_N.is_zero(n)
      then m
      else a_N.succ(
        plus(a_N.pred(n),m))
end
```

Yes, it is a signature and therefore can be used below ▷. A **where**-clause refines what a signature stands for.

⁸⁵ So here, the signature is like *P* but requires that *number* in the functor's result must be equal to *a_N.number*.

And how do we make sure in **struct ... end** that this is the case? ⁸⁶

We define the type *number* to be the type *number* of the structure *a_N*'s type *number*.

Here is a signature.

```
signature J =  
  sig  
    val new_plus : (int * int) → int  
  end
```

And here is the functor *NP*.

```
functor NP(structure a_N : N_C_R  
           structure a_P : P)  
  ▷  
  J  
  =  
  struct  
    fun new_plus(x,y)  
      = a_N.reveal(  
        a_P.plus(  
          a_N.conceal(x),  
          a_N.conceal(y)))  
  end
```



Why is this definition nonsense?

Still, it is nonsense.

¹²⁹ Why oh why?

Suppose we use this functor with *nums* and *IntArith*.

```
structure NP1 =  
  NP(structure a_N = NumStruct  
     structure a_P = IntArith)
```



And that is?

¹³⁰ Now it's obvious why the definition of the functor is nonsense.

¹³¹ The function *NumStruct.conceal* would produce numbers as *nums* and *IntArith.plus* would attempt to consume those, which is nonsense.

You have reached the end of your introduction to computation with types and functions. While computation has been popularized over the past few years, especially by the Web and consumer software, it also has a profound, intellectually challenging side. If you wish to delve deeper into this side of computing, starting from a typed viewpoint, we recommend the following tour:

References

1. Heijenoort. *From Frege to Goedel: A Source Book in Mathematical Logic, 1879–1931*. Harvard Press, 1967.
2. Pierce. *Basic Category Theory for Computer Scientists*. MIT Press, 1991.
3. Girard, Taylor, and Lafont. *Proofs and Types*. Cambridge University Press, 1989.
4. Enderton. *A Mathematical Introduction to Logic*. Academic Press, 1972.
5. Constable et alii. *Implementing Mathematics With the Nuprl Proof Development System*. Prentice Hall, 1986.

If you then wish to explore the definition of ML, you may wish to study:

1. Milner, Tofte, Harper, and MacQueen. *The Definition of Standard ML—Revised*. MIT Press, 1997.
2. Milner and Tofte. *Commentary on Standard ML*. MIT Press, 1991.

Index

add_a_steak, [47](#), [48](#), [54](#)

base, [126](#)

bool_or_int, [91](#)

box, [133](#)

chain, [95](#)

chain_item, [100](#), [101](#)

combine, [123](#)

combine_c, [124](#)

combine_s, [126](#), [127](#)

contains_fruit, [76](#)

dessert, [45](#)

divides_evenly, [98](#)

eq_fish, [64](#)

eq_fruit, [80](#)

eq_fruit_in_atom, [86](#)

eq_main, [50](#), [51](#)

eq_num, [70](#)

eq_orapl, [109](#)

*fib*s, [104](#)

*fib*s_1, [105](#)

*fib*s_2, [106](#)

find, [143](#)

fish, [57](#), [60](#)

flat_only, [74](#)

fruit, [74](#)

has_steak, [52](#), [53](#)

height, [79](#)

help, [93](#)

hot_maker, [92](#)

identity, [91](#)

in_range, [114](#)

in_range_11_16, [118](#)

in_range_c, [119](#)

in_range_c_11_16, [120](#)

IntArith, [159](#), [163](#), [165](#)

IntArith2, [167](#)

IntStruct, [157](#), [163](#)

IntStruct2, [167](#)

ints, [96](#)

is_15, [116](#)

is_bacon, [134](#)

is_mod_5_or_7, [98](#)

is_prime, [102](#)

is_vegetarian, [21](#)

is_veggie, [23](#), [25](#)

is_zero, [151](#), [152](#)

J, [174](#)

larger_of, [78](#)

less_than_15, [78](#), [117](#)

list, [109](#)

list_item, [142](#)

main, [45](#)

make_cons, [127](#), [128](#)

meza, [45](#)

N, [154](#)

new_plus, [173](#)

N_C_R, [161](#)

No_bacon, [135](#)

NP, [174](#), [175](#)

NP1, [174](#)

NPStruct, [175](#), [176](#)

num, [4](#), [152](#)

NumArith, [160](#), [163](#), [166](#)

NumArith2, [169](#)

NumStruct, [157](#), [163](#)

NumStruct2, [169](#)

NumberAsInt, [156](#), [162](#)

NumberAsInt2, [167](#)

NumberAsNum, [156](#), [162](#)

NumberAsNum2, [169](#)

occurs, [81](#)

occurs_in_sexp, [84](#)

occurs_in_slist, [84](#)

only_onions, [12–14](#)

open_faced_sandwich, [6](#)

orapl, [109](#)

Out_of_range, [142](#)

P, [158](#)
path, [147](#)
pizza, [31](#), [57](#)
plate, [22](#), [25](#)
plus, [151](#), [152](#)
PON, [158](#), [165](#)
pred, [151](#), [152](#)
prefix_123, [129](#)
[prefix_23](#), [129](#)
[prefix_3](#), [129](#)
prefixer_123, [125](#)
primes, [103](#)

rem_anchovy, [59](#)
rem_fish, [62](#), [63](#), [65](#)
rem_from_slist, [85–88](#)
rem_int, [68](#)
rem_tuna, [59–61](#)
remove_anchovy, [32](#), [33](#), [37](#), [38](#), [49](#)
rod, [22](#), [25](#)

[S](#), [171](#)
salad, [45](#)
Same, [171](#)
seasoning, [4](#)
sexp, [82](#)
shish, [22](#)
shish_kebab, [11](#)
SimIntNum, [172](#)

SimNumInt, [172](#)
SimNumNum, [173](#)
skips, [97](#)
slist, [82](#)
some_ints, [98](#)
split_only, [75](#)
subst, [110](#)
subst_anchovy_by_cheese, [40](#), [41](#)
subst_c, [122](#)
[subst_c_in_range_11_16](#), [123](#)
subst_fish, [69](#)
subst_in_sexp, [84](#)
subst_in_slist, [84](#)
subst_in_tree, [80](#)
subst_int, [69](#), [109](#)
subst_orapl, [109](#)
subst_pred, [115](#)
succ, [151](#), [152](#)

T, [177](#)
TON, [177](#)
Too_small, [151](#), [152](#)
top_anchovy_with_cheese, [38](#), [39](#)
tree, [74](#)
true_maker, [91](#)

waiting_prefix_123, [125](#)
what_bottom, [27](#), [29](#)
where_is, [134](#), [135](#)

This is for the loyal Schemers.

```
signature Ysig  
=  
sig  
  val Y :  
     $((\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow \alpha)) \rightarrow (\alpha \rightarrow \alpha)$   
end
```

```
functor Yfunc()  
▷  
  Ysig  
  =  
  struct  
    datatype  $\alpha$  T = into of  $\alpha$  T  $\rightarrow \alpha$   
    fun Y(f)  
      = H(f)(into(H(f)))  
    and H(f)(a)  
      = f(G(a))  
    and G(into(a))(x)  
      = a(into(a))(x)  
  end
```

```
structure Ystruct  
  = Yfunc()
```

No, we wouldn't forget factorial.

```
fun mk_fact(fact)(n)  
  = if (n = 0)  
    then 1  
    else n * fact(n - 1)
```

What is the value of
Ystruct.Y(*mk_fact*)(10)?

The Little MLer

Matthias Felleisen and Daniel P. Friedman

Foreword by Robin Milner

Drawings by Duane Bibby

Matthias Felleisen and Daniel Friedman are well known for gently introducing readers to difficult ideas. *The Little MLer* is an introduction to thinking about programming and the ML programming language. The authors introduce those new to programming, as well as those experienced in other programming languages, to the principles of types, computation, and program construction. Most important, they help the reader to think recursively with types about programs.

Matthias Felleisen is Professor of Computer Science at Rice University. Daniel P. Friedman is Professor of Computer Science at Indiana University. They are the authors of *The Little Schemer*, *The Seasoned Schemer*, and *A Little Java, A Few Patterns*.

The MIT Press

Massachusetts Institute of Technology • Cambridge, Massachusetts 02142

<http://mitpress.mit.edu> • FELLP 0-262-56114-X

